Problem 1.2-7: Measuring Contact Resistance

You have designed the experimental apparatus shown in Figure P1.2-7 to measure contact resistance. Four thermocouples (labeled TC_1 through TC_4) are embedded in two sample blocks at precise locations. The thermocouples are placed L_1 = 0.25 inch from the edges of the sample blocks and L_2 = 1.0 inch apart, as shown. Heat is applied to the top of the apparatus and removed from the bottom using a flow of coolant. The sides of the sample blocks are insulated. The sample blocks are fabricated from an alloy with a precisely-known and nearly constant thermal conductivity, k_s = 2.5 W/m-K. The apparatus is activated and allowed to reach steady state. The temperatures recorded by the thermocouples are TC_1 = 53.3°C, TC_2 = 43.1°C, TC_3 = 22.6°C, and TC_4 = 12.3°C. The contact resistance of interest is the interface between the sample blocks.



Figure P1.2-7: Experimental device to measure contact resistance.

a.) Use the data provided above to compute the measured heat flux in the upper and lower sample blocks.

The input parameters are entered in EES:

```
"P1.2-7 "
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

k_s=2.5 [W/m-K]                          "conductivity"
L_1=0.25 [inch]*convert(inch,m)          "distance between sensor and interface"
L_2=1.0 [inch]*convert(inch,m)           "distance between sensors"
TC_1=converttemp(C,K,53.3)               "thermocouple 1 measurement"
TC_2=converttemp(C,K,43.1)               "thermocouple 2 measurement"
TC_3=converttemp(C,K,22.6)               "thermocouple 3 measurement"
TC_4=converttemp(C,K,12.3)               "thermocouple 4 measurement"
```

The heat transfer through the sample blocks is one-dimensional, steady state conduction through a constant cross-sectional area and therefore the heat flux through the upper and lower sample blocks are given by:

$$\dot{q}_1'' = k_s \frac{(TC_1 - TC_2)}{L_2} \tag{1}$$

$$\dot{q}_2'' = k_s \frac{(TC_3 - TC_4)}{L_2} \tag{2}$$

```
q_flux_1=(TC_1-TC_2)*k_s/L_2                    "heat flux in hot block"
q_flux_2=(TC_3-TC_4)*k_s/L_2                    "heat flux in cold block"
```

The heat flux measurements are $\dot{q}_1''$ =1004 W/m^2 and $\dot{q}_2''$ =1014 W/m^2. Note that these values should be the same but are different due to measurement uncertainty or heat loss through the insulation.

b.) Use the data to compute the temperature on the hot and cold sides of the interface.

Figure 2 illustrates the measured temperatures as a function of position; the temperatures on the hot and cold sides of the interface (T_h and T_c) can be obtained by extrapolating the temperature gradient to the interface, as shown in Figure 2.



Figure 2: Measured temperatures as a function of position and extrapolated temperatures at the interface.

The temperatures at the hot and cold sides of the interface are estimated according to:

$$T_h = TC_2 - (TC_1 - TC_2)\frac{L_1}{L_2} \tag{3}$$

$$T_c = TC_3 + \left(TC_3 - TC_4\right)\frac{L_1}{L_2} \tag{4}$$

```
T_h=TC_2-(TC_1-TC_2)*L_1/L_2          "extrapolated temperature at the hot interface"
T_c=TC_3+(TC_3-TC_4)*L_1/L_2          "extrapolated temperature at the cold interface"
```

The extrapolated temperatures at the interface are T_h = 313.7 K and T_c = 298.3 K.

c.) Use the data to compute the measured contact resistance.

The average of the two heat flux measurements is:

$$\dot{q}'' = \frac{\left(\dot{q}_1'' + \dot{q}_2''\right)}{2} \tag{5}$$

The measured value of the contact resistance is therefore:

$$R_c'' = \frac{\left(T_h - T_c\right)}{\dot{q}''} \tag{6}$$

```
q_flux=(q_flux_1+q_flux_2)/2                      "average of heat flux calculations"
R_contact=(T_h-T_c)/q_flux                        "measured contact resistance"
```

The measured contact resistance is R_c'' = 0.0152 K-m^2/W.

It is important to estimate the uncertainty in your measurement. The uncertainty in the distance measurements is δL= 0.01 inch, the uncertainty in the conductivity of the sample blocks is δk_s = 0.4 W/m-K, and the uncertainty in the temperature measurements is δT = 0.5 K.

d.) Estimate the uncertainty in the measurement of the heat flux in the upper sample block, the answer for (a), manually; that is carry out the uncertainty propagation calculations explicitly.

The uncertainties are entered in EES:

```
dk_s=0.1 [W/m-K]                      "uncertainty in conductivity"
dL=0.01 [inch]*convert(inch,m)        "uncertainty in position measurements"
dT=0.5 [K]                            "uncertainty in temperature measurement"
```

The uncertainty in $\dot{q}_1''$ is related to the uncertainty in the measured quantities used to calculate $\dot{q}_1''$:

$$\dot{q}_1'' = k_s \frac{\left(TC_1 - TC_2\right)}{L_2} \tag{7}$$

The uncertainty in $\dot{q}_1''$ due to TC_1 ($\delta\dot{q}_{1,TC_1}''$) is obtained according to:

$$\delta \dot{q}''_{1,TC_1} = \frac{\partial \dot{q}''_1}{\partial TC_1}\delta T = k_s \frac{\delta T}{L_2} \tag{8}$$

and the uncertainty in $\dot{q}''_1$ due to TC_2 ($\delta \dot{q}''_{1,TC_2}$) is also:

$$\delta \dot{q}''_{1,TC_2} = k_s \frac{\delta T}{L_2} \tag{9}$$

The uncertainty in $\dot{q}''_1$ due to k_s ($\delta \dot{q}''_{1,k_s}$) is:

$$\delta \dot{q}''_{1,k_s} = \left(TC_1 - TC_2\right)\frac{\delta k_s}{L_2} \tag{10}$$

and the uncertainty in $\dot{q}''_1$ due to L_2 ($\delta \dot{q}''_{1,L_2}$) is:

$$\delta \dot{q}''_{1,L_2} = k_s \left(TC_1 - TC_2\right)\frac{\delta L}{L_2^2} \tag{11}$$

The total uncertainty in the heat flux is obtained by combining these contributions using the root-sum-square (RSS) technique:

$$\delta \dot{q}''_1 = \sqrt{\delta \dot{q}''^2_{1,TC_1} + \delta \dot{q}''^2_{1,TC_2} + \delta \dot{q}''^2_{1,k_s} + \delta \dot{q}''^2_{1,L_2}} \tag{12}$$

```
"Manual calculation of the uncertainty"
dq_flux_1_TC_1=dT*k_s/L_2                          "uncertainty in heat flux 1 due to TC_1"
dq_flux_1_TC_2=dT*k_s/L_2                          "uncertainty in heat flux 1 due to TC_2"
dq_flux_1_k_s=(TC_1-TC_2)*dk_s/L_2                 "uncertainty in heat flux 1 due to k_s"
dq_flux_1_L_2=(TC_1-TC_2)*k_s*dL/L_2^2             "uncertainty in heat flux 1 due to L_2"
dq_flux_1=sqrt(dq_flux_1_TC_1^2+dq_flux_1_TC_2^2+dq_flux_1_k_s^2+dq_flux_1_L_2^2)
        "uncertainty in heat flux 1 measurement"
```

The total uncertainty in the heat flux is $\delta \dot{q}''_1$ = 81.0 W/m^2.

e.) Verify that EES' uncertainty propagation function provides the same answer obtained in (d).

The uncertainty propagation capability of EES is accessed by selecting Uncertainty Propagation from the Calculate menu (Figure 3).



Figure 3: Uncertainty Propagation Window.

The calculated variable of interest is q_flux_1 and this should be selected from the Calculated variable list. The measured variables with uncertainty include the variables k_s, L_1, L_2, TC_1, TC_2, TC_3, and TC_4; these should be selected from the Measured variable list. The uncertainty associated with these measured variables can be specified by selecting Set uncertainties (Figure 4).



Figure 4: Uncertainties of Measured Variables Window.

The absolute uncertainties of each of the measured variables are assigned using the corresponding variable names (Figure 4). Select OK twice to see the results of the uncertainty propagation calculation (Figure 5).



Figure 5: Uncertainty Results Window.

Notice that the heat flux uncertainty calculated by EES is also 81.0 W/m^2. The Uncertainty Results Window also delineates the sources of the uncertainty.

f.) Use EES' uncertainty propagation function to determine the uncertainty in the measured value of the contact resistance. What is the % uncertainty in your measurement?

Rather than the variable q_flux_1, the variable R_contact is selected in the Uncertainty of Measured Variables Window. The result of the calculation is shown in Figure 6.



Figure 6: Uncertainty Results Window.

The uncertainty in the contact resistance is 0.0017 K-m^2/W or 11%.

g.) Which of the fundamental measurements that are required by your test facility should be improved in order to improve your measurement of the contact resistance? That is, would you focus your attention on reducing δk_s, δL, or δT? Justify your answer.

Examination of Figure 6 suggests that the uncertainty in the contact resistance is due almost entirely to the temperature measurements. Therefore, I would focus my attention on reducing δT.