PROBLEM 1.4-2 (1-10 in text): Mass Flow Meter (revisited)

Reconsider the mass flow meter that was investigated in Problem 1.3-9 (1-9 in text). The conductivity of the material that is used to make the test section is not actually constant as was assumed in Problem 1-9 but rather depends on temperature according to:

$$k = 10\frac{\text{W}}{\text{m-K}} + 0.035\left[\frac{\text{W}}{\text{m-K}^2}\right]\left(T - 300\left[\text{K}\right]\right)$$

a.) Develop a numerical model of the mass flow meter using EES. Plot the temperature as a function of radial position for the conditions shown in Figure P1.3-9 (Figure P1-9 in the text) with the temperature-dependent conductivity.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
r_out=1.0 [inch]*convert(inch,m)                "outer radius of measurement section"
r_in=0.75 [inch]*convert(inch,m)                "inner radius of measurement section"
h_bar_out=10 [W/m^2-K]                          "external convection coefficient"
T_infinity_C=20 [C]                             "ambient temperature in C"
T_infinity=converttemp(C,K,T_infinity_C)        "ambient temperature"
T_f=converttemp(C,K, 18 [C])                    "fluid temperature"
g```=1e7 [W/m^3]                                "volumetric rate of thermal energy generation"
m_dot=0.75 [kg/s]                               "mass flow rate"
th_ins=0.25 [inch]*convert(inch,m)              "thickness of insulation"
k_ins=1.5 [W/m-K]                               "insulation conductivity"
L= 3 [inch]*convert(inch,m)                     "length of test section"
C=2500 [W/m^2-K]                                "constant for convection relationship"
h_bar_in=C*(m_dot/1 [kg/s])^0.8                 "internal convection coefficient"
```

A function is defined that returns the conductivity of the material:

```
Function k_t(T)
    "This function returns the conductivity of the test section material as a function of temperature"
    k_t=10 [W/m-K]+0.035 [W/m-K^2]*(T-300 [K])
end
```

A uniform distribution of nodes is used, the radial location of each node (r_i) is:

$$r_i = r_{in} + \frac{(i-1)}{(N-1)}\left(r_{out} - r_{in}\right) \quad \text{for } i = 1..N \tag{1}$$

where N is the number of nodes. The radial distance between adjacent nodes (Δr) is:

$$\Delta r = \frac{\left(r_{out} - r_{in}\right)}{\left(N-1\right)} \tag{2}$$

```
N=51 [-]                                  "number of nodes"
DELTAr=(r_out-r_in)/(N-1)                 "distance between adjacent nodes (m)"
"Set up nodes"
duplicate i=1,N                           "this loop assigns the radial location to each node"
    r[i]=r_in+(r_out-r_in)*(i-1)/(N-1)
end
```

An energy balance is carried out on a control volume surrounding each node. For node 1, placed at the inner surface (Figure P1.4-2-1):

$$\dot{q}_{conv,in} + \dot{q}_{outer} + \dot{g} = 0 \tag{3}$$



Figure P1.4-2-1: Control volume around node 1.

The rate equation for convection is:

$$\dot{q}_{conv,in} = \bar{h}_{in}\, 2\,\pi\, r_{in}\, L\left(T_f - T_1\right) \tag{4}$$

The rate equation for conduction is:

$$\dot{q}_{outer} = k_{T=\left(T_1+T_2\right)/2}\, 2\,\pi\left(r_{in} + \frac{\Delta r}{2}\right) L \frac{\left(T_2 - T_1\right)}{\Delta r} \tag{5}$$

The rate equation for generation is:

$$\dot{g} = 2\,\pi\, r_{in} \frac{\Delta r}{2} L\, \dot{g}''' \tag{6}$$

```
"Node 1"
q_dot_conv_in=h_bar_in*2*pi*r_in*L*(T_f-T[1])                               "convection from fluid"
g_dot[1]=2*pi*r_in*L*DELTAr*g```/2                                          "generation"
q_dot_outer[1]=k_t((T[1]+T[2])/2)*2*pi*(r[1]+DELTAr/2)*L*(T[2]-T[1])/DELTAr "conduction from node 2"
q_dot_conv_in+q_dot_outer[1]+g_dot[1]=0                                     "energy balance on node 1"
```

An energy balance on an internal node is shown in Figure P1.4-2-2:

$$\dot{q}_{inner} + \dot{q}_{outer} + \dot{g} = 0 \tag{7}$$



Figure P1.4-2-2: Control volume around internal node *i*.

The rate equations for conduction are:

$$\dot{q}_{outer} = k_{T=(T_i+T_{i+1})/2}\, 2\,\pi \left(r_{in} + \frac{\Delta r}{2} \right) L \frac{\left(T_{i+1} - T_i\right)}{\Delta r} \tag{8}$$

$$\dot{q}_{inner} = k_{T=(T_i+T_{i-1})/2}\, 2\,\pi \left(r_{in} - \frac{\Delta r}{2} \right) L \frac{\left(T_{i-1} - T_i\right)}{\Delta r} \tag{9}$$

The rate equation for generation is:

$$\dot{g} = 2\,\pi\, r_i\, \Delta r\, L\, \dot{g}''' \tag{10}$$

```
"Internal nodes"
duplicate i=2,(N-1)
     q_dot_inner[i]=k_t((T[i]+T[i-1])/2)*2*pi*(r[i]-DELTAr/2)*L*(T[i-1]-T[i])/DELTAr
                                                                           "conduction from inner node"
     q_dot_outer[i]=k_t((T[i]+T[i+1])/2)*2*pi*(r[i]+DELTAr/2)*L*(T[i+1]-T[i])/DELTAr
                                                                           "conduction from outer node"
     g_dot[i]=2*pi*r[i]*L*DELTAr*g```                                      "generation"
     q_dot_inner[i]+q_dot_outer[i]+g_dot[i]=0                              "energy balance on node i"
end
```

An energy balance on node *N* placed on the outer surface is shown in Figure P1.4-2-3:

$$\dot{q}_{inner} + \dot{q}_{air} + \dot{g} = 0 \tag{11}$$



Figure P1.4-2-3: Control volume around internal node *N*.

The rate equation for the heat transfer with the air is:

$$\dot{q}_{air} = \frac{\left(T_\infty - T_N\right)}{\left(R_{ins} + R_{conv,out}\right)} \tag{12}$$

where

$$R_{ins} = \frac{\ln\left[\frac{\left(r_{out} + th_{ins}\right)}{r_{out}}\right]}{2\,\pi\,L\,k_{ins}} \tag{13}$$

$$R_{conv,out} = \frac{1}{2\,\pi\left(r_{out} + th_{ins}\right)L\,\bar{h}_{out}} \tag{14}$$

The rate equation for conduction is:

$$\dot{q}_{inner} = k_{T=\left(T_N + T_{N-1}\right)/2}\,2\,\pi\left(r_{out} - \frac{\Delta r}{2}\right)L\frac{\left(T_{N-1} - T_N\right)}{\Delta r} \tag{15}$$

The rate equation for generation is:

$$\dot{g} = 2\,\pi\,r_{out}\,\frac{\Delta r}{2}\,L\,\dot{g}''' \tag{16}$$

```
"Node N"
R_ins=ln((r_out+th_ins)/r_out)/(2*pi*L*k_ins)            "resistance to conduction through insulation"
R_conv_out=1/(2*pi*(r_out+th_ins)*L*h_bar_out)           "resistance to convection from outer surface"
q_dot_air=(T_infinity-T[N])/(R_ins+R_conv_out)           "heat transfer from air"
q_dot_inner[N]=k_t((T[N]+T[N-1])/2)*2*pi*(r_out-DELTAr/2)*L*(T[N-1]-T[N])/DELTAr
                                                         "conduction from node N-1"
g_dot[N]=2*pi*r_out*L*DELTAr*g```/2                      "generation"
q_dot_air+q_dot_inner[N]+g_dot[N]=0                      "energy balance on node N"
```

The solution is converted to degrees Celsius:

```
duplicate i=1,N
    T_C[i]=converttemp(K,C,T[i])                         "convert solution to deg. C"
end
```

The solution is illustrated in Figure P1.4-2-4.



Figure P1.4-2-4: Temperature as a function of radius.

b.) Verify that your numerical solution limits to the analytical solution from Problem 1.3-9 (1-9 in the text) in the limit that the conductivity is constant.

The conductivity function is modified temporarily so that it returns a constant value:

```
Function k_t(T)
    "This function returns the conductivity of the test section material as a function of temperature"
    k_t=10 [W/m-K]{+0.035 [W/m-K^2]*(T-300 [K])}
end
```

The analytical solution from P1.3-9 is programmed and used to compute the analytical solution at each node:

```
"Analytical solution from P1.3-9"
k=k_t(300 [K])                                  "conductivity to use in the solution"
T_r_out=-g```*r_out^2/(4*k)+C_1*ln(r_out)+C_2    "temperature at outer surface of section"
dTdr_r_out=-g```*r_out/(2*k)+C_1/r_out           "temperature gradient at outer surface of section"
-k*2*pi*r_out*L*dTdr_r_out=(T_r_out-T_infinity)/(R_ins+R_conv_out) "boundary condition at r=r_out"
T_r_in=-g```*r_in^2/(4*k)+C_1*ln(r_in)+C_2       "temperature at inner surface of section"
dTdr_r_in=-g```*r_in/(2*k)+C_1/r_in              "temperature gradient at inner surface of section"
h_bar_in*2*pi*r_in*L*(T_f-T_r_in)=-k*2*pi*r_in*L*dTdr_r_in  "boundary condition at r=r_in"

duplicate i=1,N
    T_an[i]=-g```*r[i]^2/(4*k)+C_1*ln(r[i])+C_2   "temperature"
    T_an_C[i]=converttemp(K,C,T_an[i])           "in C"
end
```

Figure P1.4-2-5 illustrates the temperature distribution predicted by the numerical and analytical solutions in the limit that *k* is constant.



Figure P1.4-2-5: Temperature as a function of radius predicted by the analytical and numerical models in the limit that *k* is constant.

c.) What effect does the temperature dependent conductivity have on the calibration curve that you generated in part (d) of Problem 1.3-9 (1-9)?

The quantity measured by the meter is the difference between the temperature at the center of the pipe wall (T[26] when 51 nodes are used) and the fluid temperature:

```
DT=T[26]-T_f                "temperature difference"
```

Figure P1.4-2-6 illustrates the calibration curve (i.e., the relationship between the temperature difference and the mass flow rate) with and without the temperature dependent conductivity included.



Figure P1.4-2-6: Calibration curve generated with and without the temperature dependent conductivity included.