
Problem 1.7-5 
A flux meter is illustrated in Figure P1.7-5. 

Figure P1.7-5: Flux meter. 

 
A thin plate is clamped on either end to a casing that is maintained at Tc = 20ºC.  There is a 
contact resistance between the plate and the casing, R′′ = 1x10-4 K-m2/W.  The thickness of the 
plate is th = 1 mm and the half-width of the plate is L = 2 cm.  The plate conductivity is k = 10 
W/m-K.  The back of the plate is insulated and the front of the plate is mounted within an 
evacuated enclosure in order to eliminate convection.  You may model radiation loss from the 
plate surface using an effective "radiation" heat transfer coefficient, calculated according to: 

34radh Tε σ≈ , where T is the average absolute temperature of the plate and the surroundings 
(take T = 300 K for this problem).  The plate radiates to surroundings at T∞ = 20ºC.  The 
nominal flux on the plate is q′′ = 900 W/m2.  You may assume that the plate temperature 
distribution is 1-D in the x-direction.  You may also assume that the emissivity of the plate, ε, is 
one and therefore all of the flux on the plate is absorbed.  The flux meter operates by correlating 
the difference between the temperature at the center of the plate and the casing with the applied 
heat flux. 
a.) Derive the governing differential equation that governs the temperature within the plate.  

Clearly show your steps. 
 
An energy balance on a differential control volume (see Figure 2) leads to: 
 
 ( )x x dx radq q dxW q h dxW T T+ ∞′′+ = + −  (1)  

Figure 2: Differential energy balance. 

 
Expanding Eq. (1) leads to: 
 



 ( )x x rad
dqq q dxW q dx h dxW T T
dx ∞′′+ = + + −  (2)  

 
Substituting Fourier's law into Eq. (2) leads to: 
 

 ( )rad
d dTq dxW kW th dx h dxW T T
dx dx ∞

⎛ ⎞′′ = − + −⎜ ⎟
⎝ ⎠

 (3)  

 
or 

2
rad radh hd T qT T

dx k th k th k th ∞

′′
− = − −  (4)  

 
b.) What are the boundary conditions for the differential equation? 
 
An interface energy balance at x = 0 leads to: 
 

 ( )0

0

c x

x

T T dTk
R dx

=

=

− ⎛ ⎞= − ⎜ ⎟′′ ⎝ ⎠
 (5)  

 
and an interface energy balance at x = L leads to: 
 

 0
x L

dT
dx =

⎛ ⎞ =⎜ ⎟

 (6)  

 
c.) Determine the solution to the differential equation from (a) subject to the boundary 

conditions from (b) without using Maple.   
 
The solution to Eq. (4) is split into a homogeneous and particular solution: 
 
 h pT T T= +  (7)  
 
Substituting Eq. (7) into Eq. (4) leads to: 

ph rad rad rad
h p

d Td T h h hqT T T
dx k th dx k th k th k th ∞

=

′′
− + − = − −  (8)  

 
The solution to the homogeneous differential equation: 
 

 
2

2 0h rad
h

d T h T
dx k th

− =  (9)  



 
is an exponential: 
 
 ( )exphT C m x=  (10)  
 
Substituting Eq. (10) into Eq. (9) leads to: 
 

 ( ) ( )2 exp exp 0radhC m m x C m x
k th

− =  (11)  

 
which is solved by: 
 

 2 radhm
k th

=  (12)  

 
Therefore, there are two solutions corresponding to the two roots of Eq. (12): 
 
 ( ) ( )1 2exp exphT C m x C m x= + −  (13)  
 
where C1 and C2 are undetermined constants and m is: 
 

 radhm
k th

=  (14)  

 
The solution to the particular differential equation: 
 

 
2

2
p rad rad

p

d T h hqT T
dx k th k th k th ∞

′′
− = − −  (15)  

 
is, by inspection: 
 

 p
rad

qT T
h ∞

′′
= +  (16)  

 
Substituting Eqs. (13) and (16) into Eq. (7) leads to: 
 

 ( ) ( )1 2exp exp
rad

qT C m x C m x T
h ∞

′′
= + − + +  (17)  

 
Substituting Eq. (17) into Eq. (5) leads to: 
 



 ( )
1 2

1 2

c
rad

qT C C T
h

k C m C m
R

∞

⎛ ⎞′′
− − − −⎜ ⎟

⎝ ⎠ = − −
′′

 (18)  

 
Substituting Eq. (17) into Eq. (6) leads to: 
 
 ( ) ( )1 2exp exp 0C m mL C m mL− − =  (19)  
 
d.) Plot the temperature in the plate as a function of position. 
 
The inputs are entered in EES: 
 
"P1.7-5" 
$UnitSystem SI MASS  DEG PA C J 
$Tabstops 0.2 0.4 0.6 0.8 3.5 
 
"Inputs" 
q_flux=900 [W/m^2] "solar flux" 
th=1 [mm]*convert(mm,m) "thickness of plate" 
L=2 [cm]*convert(cm,m) "half-length of plate" 
k=10 [W/m-K]  "conductivity of plate" 
Rc=1e-4 [K-m^2/W] "contact resistance at clamped edges" 
T_c=converttemp(C,K,20[C]) "casing temperature" 
T_infinity=converttemp(C,K,20[C]) "surrounding temperature" 
e=1 [-]    "emissivity" 
T_bar=300 [K]  "average temperature" 
h_rad=4*sigma#*e*T_bar^3 "radiation heat transfer coefficient" 
 
Equations (18) and (19) are entered in EES in order to determine the constants C1 and C2: 
 
"boundary conditions" 
m=sqrt(h_rad/(k*th)) "fin constant" 
(T_c-C_1-C_2-q_flux/h_rad-T_infinity)/Rc=-k*(C_1*m-C_2*m) "at x=0" 
C_1*m*exp(m*L)-C_2*m*exp(-m*L)=0 "at x=L"  
 
The solution is programmed in EES: 
 
"solution" 
{x_bar=0 [-]}  "dimensionless position" 
x=x_bar*L   "position" 
T=C_1*exp(m*x)+C_2*exp(-m*x)+q_flux/h_rad+T_infinity "temperature solution" 
T_Celsius=converttemp(K,C,T) "in C" 
 
Figure 3 illustrates the temperature as a function of position in the plate. 

Figure 3: Temperature as a function of position in the plate. 

 
e.) Use Maple to solve the differential equation and obtain symbolic expressions for the 

boundary conditions.  Implement the Maple expressions into EES and show that your answer 
is identical to the one determined in (c). 

 
The differential equation, Eq. (4), is entered in Maple: 
 
> restart; 
> ODE:=diff(diff(T(x),x),x)-h_rad*T(x)/(k*th)=-q_flux/(k*th)-h_rad*T_infinity/(k*th); 

 := ODE  =  − ⎛

⎝

and solved: 
 
> Ts:=dsolve(ODE); 

Symbolic expressions for the boundary conditions are obtained: 
 
> BC1:=(T_c-rhs(eval(Ts,x=0)))/Rc=-k*rhs(eval(diff(Ts,x),x=L)); 

 −  −  − T_c _C2 _C1

> BC2:=rhs(eval(diff(Ts,x),x=L))=0; 

The original equations in EES are commented out and the expressions from Maple are copied to 
EES: 
 
{"boundary conditions" 
m=sqrt(h_rad/(k*th)) "fin constant" 
(T_c-C_1-C_2-q_flux/h_rad-T_infinity)/Rc=-k*(C_1*m-C_2*m) "at x=0" 
C_1*m*exp(m*L)-C_2*m*exp(-m*L)=0 "at x=L" 
 
"solution" 
{x_bar=0 [-]}  "dimensionless position" 
x=x_bar*L   "position" 
T=C_1*exp(m*x)+C_2*exp(-m*x)+q_flux/h_rad+T_infinity "temperature solution" 
T_Celsius=converttemp(K,C,T) "in C"} 
 
"Maple solution" 
(T_c-C_2-C_1-1/h_rad*(q_flux+h_rad*T_infinity))/Rc = -k*(1/k^(1/2)/th^(1/2)*h_rad^(1/2)*& 
exp(1/k^(1/2)/th^(1/2)*h_rad^(1/2)*L)*C_2-1/k^(1/2)/th^(1/2)*h_rad^(1/2)*exp(-1/k^(1/2)/& 
th^(1/2)*h_rad^(1/2)*L)*C_1) "boundary condition at x=0" 
 
1/k^(1/2)/th^(1/2)*h_rad^(1/2)*exp(1/k^(1/2)/th^(1/2)*h_rad^(1/2)*L)*C_2-1/k^(1/2)/th^(1/2)*& 
h_rad^(1/2)*exp(-1/k^(1/2)/th^(1/2)*h_rad^(1/2)*L)*C_1 = 0 "boundary condition at x=L" 
 
The solution from Maple is copied into EES: 
 
T = exp(1/k^(1/2)/th^(1/2)*h_rad^(1/2)*x)*C_2+exp(-1/k^(1/2)/th^(1/2)*h_rad^(1/2)*x)*C_1& 
 +1/h_rad*(q_flux+h_rad*T_infinity) "solution" 
x=x_bar*L   "position" 
T_Celsius=converttemp(K,C,T) "in C" 
 
The solution is overlaid onto the plot from (d) in Figure 4. 

Figure 4: Temperature as a function of position in the plate using Maple solution. 



 
f.) Prepare a calibration curve for the flux meter - plot the heat flux as a function of the 

difference between the temperature at the center of the plate and the casing. 
 
Figure 5 illustrates the difference between the temperature at the center of the plate as a function 
of the applied heat flux. 
 

0 2.5 5 7.5 10 12.5 15 17.5 20 22.5 25
0

200

400

600

800

1000

1200

1400

Temperature rise (K)

H
ea

t f
lu

x 
(W

/m
2 )

 
Figure 5: Calibration curve for the flux meter. 

 
g.) If the uncertainty in the measurement of the temperature difference is δΔT = 0.5 K then what 

is the uncertainty in the measurement of the heat flux? 
 
The uncertainty in the measurement of the heat flux is approximately: 
 

 qq T
T

δ δ
δ

′′∂′′ = Δ
Δ

 (20)  

 

where q
Tδ
′′∂

Δ
= 55 W/m2-K according to Figure 5.  Therefore, the uncertainty in the heat flux is 

approximately 22 W/m2. 
 
 
 
 
 
 
 
 
 
 


