Problem 1.2-14: Ice Rink

Ice for an ice skating rink is formed by running refrigerant at T_r = -30°C through a series of cast iron pipes that are embedded in concrete, as shown in Figure P1.2-14. The cast iron pipes have an outer diameter of $D_{o,p}$ = 4 cm and an inner diameter of $D_{i,p}$ = 3 cm. The pipes are spaced L_{ptp} = 8.0 cm apart. The heat transfer coefficient between the refrigerant and the pipe surface is $\overline{h}_r$ = 100 W/m^2-K. The concrete slab is L_c = 8 cm thick and the pipes are in the center of the slab. The bottom of the slab is insulated (assume perfectly). The thermal conductivity of concrete and iron are k_c = 4.5 W/m-K and k_{iron} = 51 W/m-K, respectively.

An L_{fill} = 1 cm thick layer of water is placed on the top of the concrete slab. The refrigerant cools the top of the slab and the water turns to ice slowly. Assume that the water is stagnant and can be treated as a solid. The conductivity of ice and water are k_{ice} = 2.2 W/m-K and k_w = 0.6 W/m-K, respectively. The heat transfer coefficient between the top of the water layer and the surrounding air at T_a = 15°C is $\overline{h}_a$ = 10.0 W/m^2-K. The top of the water surface has an emissivity of ε = 0.90 and radiates to surroundings at T_{sur} = 15°C.



Figure P1.2-14: Schematic of ice rink

a.) Draw a network that represents this situation using 1-D resistances. (Some of the resistances must be approximate since it is not possible to exactly calculate a 1-D resistance to the conduction heat flow in the concrete). Include an energy term related to the energy that is added to the system by the generation of ice. Clearly label the resistors.

The resistance network is shown in Figure 1.



The resistors include:
R_{rad} = radiation resistance
$R_{conv,a}$ = convection resistance to air
R_w = conduction through water
R_{ice}= conduction through ice
R_c = conduction through concrete
R_p = conduction through pipe
$R_{conv,r}$ = convection resistance to refrigerant

Figure 1: Resistance network representing the ice rink

The resistance network interacts with the air temperature (T_a), the surroundings (T_s), and the refrigerant (T_r). There is a heat transfer rate at the interface between the ice and water related to the heat removed from the water in order to form more ice. This heat is accepted because there is more energy removed by the refrigerant than is provided from the surroundings.

b.) Estimate the magnitude of each of the resistances in your network when the ice is 0.5 cm thick (i.e., L_{ice} = 0.5 cm).

We will deal with a unit cell of the sub-floor structure, as shown in Figure 2:



Figure 2: Unit cell of the sub-floor structure

The solution process will be described in the context of EES. It is assumed that you have already been exposed to the EES software by carrying out the self-guided tutorial contained in Appendix A. The first step in preparing a successful solution to any problem with EES is to enter the inputs to the problem and set their units. Experience has shown that it is generally best to work exclusively in SI units (m, J, K, kg, Pa, etc.). This unit system is entirely self-consistent. If the problem statement includes parameters in other units they are converted to SI units within the "Inputs" section of the code. The upper section of your EES code should look something like:

```
$UnitSystem SI MASS RAD PA K J
```

```
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
T_r=converttemp(C,K,-30 [C])            "refrigerant temperature (K)"
h_r=100 [W/m^-K]                        "heat transfer coefficient between refrigerant and pipe"
D_i_p=3.0*convert(cm,m)                 "pipe inner diameter (m)"
D_o_p=4.0*convert(cm,m)                 "pipe outer diameter (m)"
k_iron=51 [W/m-K]                       "pipe conductivity (W/m-K)"
L_ptp=8.0*convert(cm,m)                 "pipe-to-pipe distance (m)"
k_c=4.5 [W/m-K]                         "concrete thermal conductivity (W/m-K)"
L_c=8.0*convert(cm,m)                   "thickness of concrete (m)"
L_p=4.0*convert(cm,m)                   "center of pipe to upper surface of concrete thickness (m)"
L_fill=1.0*convert(cm,m)                "thickness of water layer on concrete (m)"
k_ice=2.2 [W/m-K]                       "conductivity of ice (W/m-K)"
k_w=0.6 [W/m-K]                         "conductivity of water (W/m-K)"
h_a=10 [W/m^2-K]                        "water-to-air heat transfer coefficient (W/m^2-K)"
T_a=converttemp(C,K,15 [C])             "temperature of air on top of slab (K)"
T_sur=converttemp(C,K,15 [C])           "temperature of radiation surrounding on top (K)"
e=0.9 [-]                               "emissivity of water surface (-)"
W=1 [m]                                 "width of surface (m)"
L_ice=0.5*convert(cm,m)                 "thickness of ice (m)"
```

The radiation resistance is:

$$R_{rad} = \frac{1}{4\,\sigma\,\varepsilon\,\overline{T}^{3}\,A_s} \tag{1}$$

where

$$\overline{T} = \frac{T_{sur} + T_s}{2} \tag{2}$$

and T_s is the temperature at the surface of the water. The area of water in the unit cell that is exposed to air is:

$$A_s = L_{ptp}\,W \tag{3}$$

where W is the width of the unit cell into the page (here, W = 1 m for a solution per unit length of the floor).

The surface temperature cannot be known until the problem is solved and yet it must be used to calculate the resistance to radiation, R_{rad}. One of the nice things about using the Engineering Equation Solver (EES) software to solve this problem is that the software can deal with this type of nonlinearity and provide the solution to the implicit equation. It is this capability that simultaneously makes EES so powerful and yet sometimes, ironically, difficult to use. EES should be able to solve equations regardless of the order in which they are entered. However, you should enter equations in a sequence that allows you to solve them as you enter them; this is exactly what you would be forced to do if you were to solve the problem using a typical programming language (e.g., MATLAB, FORTRAN, etc.). This technique of entering your equations in a systematic order provides you with the opportunity to debug each subset of equations as you move along rather than waiting until you have entered all of your equations and tried to solve only to find that there are multiple problems. Another benefit of approaching a problem in this manner is that you can consistently update your guess values associated with the variables in your problem; EES solves your equations using a nonlinear relaxation technique and therefore the closer your variables are to "reasonable" values the better this process will go.

To proceed with the solution to this ice rink problem using EES, it is a helpful idea to assume initially a reasonable surface temperature (e.g., 273 K) so that it is possible to estimate the radiation resistance and continue with the solution. The next few lines in your EES code should look something like:

```
"Resistances"
"Radiation"
A_s=W*L_ptp                     "surface area"
T_s=273.2                       "this is a guess for the surface temperature - eventually we will
                                 comment this out"
T_bar=(T_s+T_sur)/2             "average temperature"
R_rad=1/(4*e*sigma#*T_bar^3*A_s)  "radiation resistance"
```

If you solve the equations that have been entered (Calculate/Solve) you can check that your answers make sense and you can verify that your equations have a consistent set of units. It would be good to do this and then update your guess values (Calculate/Update Guesses); this operation sets the guess values for each of your variables to their current value and therefore helps EES iterate to the correct solution. Finally, you should set the units for each of your variables. The best way to do this is to go to the Variable Information window (Options/Variable Info) and enter the unit for each variable in the Units column. Once you have done that you can check units (Calculate/Check Units) in order to make sure that all of the units you set are consistent with the equations that you've entered.

The convection resistance to the air is:

$$R_{conv,a} = \frac{1}{h_a\, A_s} \tag{4}$$

The conduction resistances through the water and ice are:

$$R_w = \frac{\left(L_{fill} - L_{ice}\right)}{k_w\, A} \tag{5}$$

$$R_w = \frac{L_{ice}}{k_{ice}\, A} \tag{6}$$

We will learn how to calculate the resistances of the concrete and pipe more exactly when we get to extended surfaces and 2-D conduction. For now we will estimate them very approximately using the concept of an effective length and cross-sectional area for conduction. The length for conduction will be taken to be the distance that the pipe is submerged beneath the surface and the area will be taken to be the area of the unit cell:

$$R_c = \frac{L_p}{k_c\, A_s} \tag{7}$$

The pipe resistance is taken to be the resistance of half a cylinder (in fact, the bottom of the pipe probably also participates if the pipe is very conductive):

$$R_p = \frac{\ln\left(\dfrac{D_{o,p}}{D_{i,p}}\right)}{\pi\, k_{iron}\, W} \tag{8}$$

Finally, the convection resistance to the refrigerant is:

$$R_{conv,r} = \frac{2}{h_r\, \pi\, D_{o,p}\, W} \tag{9}$$

The resulting EES code is shown below:

```
R_conv_a=1/(h_a*A_s)                        "air convection resistance"
R_w=(L_fill-L_ice)/(k_w*A_s)                "resistance of water layer"
R_ice=L_ice/(k_ice*A_s)                     "resistance of ice layer"
R_c=L_p/(k_c*A_s)                           "concrete resistance (approximate)"
R_p=2*ln(D_o_p/D_i_p)/(2*pi*W*k_iron)       "pipe resistance"
R_conv_r=1/(h_r*W*pi*D_i_p/2)               "refrigerant convection resistance"
```

EES will calculate the resistances in the network (although the radiation resistance continues to be only approximate). These resistances are placed on the network in Figure 3.



Figure 3: Resistances calculated for ice rink

Resistance networks often provide substantial intuition relative to a problem. For example, Figure 3 shows that the resistances associated with convection and radiation from the surface of the water are of the same order of magnitude and large relative to others in the circuit; therefore, both radiation and convection is important for this problem. If the radiation resistance had been much larger than the convection resistance (as is often the case in forced convection problems where the convection coefficient is much larger) then radiation could be neglected; the smallest resistance in a parallel network will dominate the problem because most of the thermal energy will tend to flow through that resistance.

In a series resistance network, the larger resistors dominate the problem and the smaller ones can be neglected. Therefore, we could safely neglect the conduction resistance through the water as it is small relative to the parallel combination of the radiation and convection resistances. Similarly, conduction through the ice and the pipe are not important to this problem. It is almost always a good idea to estimate the size of the resistances in a heat transfer problem prior to solving it; often it is possible to simplify the problem considerably and the size of the resistances can certainly be used to guide your efforts. For the ice rink problem, a detailed analysis of conduction through the pipe would be a misguided use of your time whereas a more accurate simulation of the conduction through the concrete would be very important.

c.) Calculate the rate of change in the thickness of the ice when the ice thickness is 0.5 cm.

The heat transfer from the ice/water interface to the refrigerant ($\dot{q}_r$) is higher than the heat transfer from the air and surroundings to the ice/water interface ($\dot{q}_a$) and therefore ice will be formed. These heat transfer rates can be estimated according to:

$$\dot{q}_r = \frac{\left(T_{ice} - T_r\right)}{R_{ice} + R_c + R_p + R_{conv,r}} \tag{10}$$

$$\dot{q}_a = \frac{\left(T_a - T_{ice}\right)}{R_w + \left[\dfrac{1}{R_{rad}} + \dfrac{1}{R_{conv,a}}\right]^{-1}} \tag{11}$$

An energy balance at the interface leads to:

$$\dot{q}_r - \dot{q}_a = A\, h_{fs} \frac{dL_{ice}}{dt} \tag{12}$$

where h_{fs} is the latent heat of fusion for ice and $\frac{dL_{ice}}{dt}$ is the rate of ice formation. The additional EES code needed to solve this problem is:

```
"Rate of ice formation"
h_fs=Enthalpy_fusion(Water)                    "enthalpy of fusion of ice"
rho_ice=1000                                   "density of ice"
T_ice=convertTemp(C,K,0)                       "temperature at which water freezes"

q_dot_r=(T_ice-T_r)/(R_conv_r+R_p+R_c+R_ice)          "heat transfer to refrigerant"
q_dot_a=(T_a-T_ice)/(R_w+(1/R_conv_a+1/R_rad)^(-1))  "heat transfer from air"
```

```
dLicedt=(q_dot_r-q_dot_a)/(A_s*rho_ice*h_fs)                "rate of change of ice layer"
```

At this point, we can use the heat transfer rates to recalculate the water surface temperature (as opposed to assuming it).

$$T_s = T_{ice} + \dot{q}_a R_w \quad (13)$$

It is necessary to comment out or delete the equation that provided the assumed surface temperature and instead calculate the surface temperature correctly.

```
{T_s=273.2}             "this is a guess for the surface temperature - eventually we will comment this out"
T_s=T_ice+q_dot_a*R_w          "recalculate the surface temperature to make radiation resistance exact"
```

The rate of formation of ice is 2.6e-6 m/s (or 0.94 cm/hr). Note that it will take about 1 hr to freeze all of the water on the rink based on this answer; the rate of ice formation will not be significantly affected by the amount of ice because the conduction resistances of the ice and water were found to be relatively insignificant. Once the ice is completely frozen, the surface temperature of the ice will drop until $\dot{q}_r$ is balanced by $\dot{q}_a$.