Problem 1.2-19

Figure P1.2-19 illustrates a cross-section of a thermal protection suit that is being designed for an astronaut.



Figure P1.2-19: Cross-section of thermal protection suit.

The suit consists of a liner that is immediately adjacent to the skin. The skin temperature is maintained at T_b = 37ºC by the flow of blood in the tissue. The liner is th_{liner} = 1 cm thick and has conductivity k_{liner} = 0.06 W/m-K. A thin heater is installed at the outer surface of the liner. Outside of the heater is a layer of insulation that is th_{ins} = 4 cm with conductivity k_{ins} = 0.06 W/m-K. Finally, the outer layer of the suit is th_{ext} = 1 mm thick with conductivity k_{ext} = 14.5 W/m-K. The outer surface of the external layer has emissivity ε = 0.25 and is exposed by radiation only to outer space at T_{space} = 4 K.

a.) You want to design the heater so that it completely eliminates any heat loss from the skin. What is the heat transfer per unit area required?

The inputs are entered in EES and the units converted to base SI units:

```
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
T_b=converttemp(C,K,37 [C])                     "tissue temperature"
th_ins=4 [cm]*convert(cm,m)                     "insulation thickness"
k_ins=0.06 [W/m-K]                              "insulation conductivity"
th_ext=1 [mm]*convert(mm,m)                     "exterior wall thickness"
k_ext=14.5 [W/m-K]                              "conductivity of exterior wall"
emm=0.25 [-]                                    "emissivity of exterior wall"
T_space = 4 [K]                                 "temperature of space"
th_liner=1 [cm]*convert(cm,m)                   "liner thickness"
k_liner=0.06 [W/m-K]                            "liner conductivity"
A=1 [m^2]                                       "do problem on a unit area basis"
```

The units of each variable are set by right-clicking on each variable in the Solution Window and setting the units in the Units dialog (Figure 2):



Figure 2: Set units for variables.

The units are checked by selecting Check Units from the Calculate menu. A resistance diagram that represents the suit is shown in Figure 3.



Figure 3: Resistance network representation of space suit.

The resistance to conduction through the liner, insulation and external layer are computed according to:

$$R_{liner} = \frac{th_{liner}}{k_{liner}\ A} \tag{1}$$

$$R_{ins} = \frac{th_{ins}}{k_{ins}\ A} \tag{2}$$

$$R_{ext} = \frac{th_{ext}}{k_{ext}\ A} \tag{3}$$

where A is taken to be 1 m^2 to do the problem on a per unit area basis.

```
"part (a)"
R_cond_liner=th_liner/(k_liner*A)      "resistance to conduction through liner"
R_cond_ins=th_ins/(k_ins*A)            "resistance to conduction through the insulation"
R_cond_ext=th_ext/(k_ext*A)            "resistance to conduction through exterior wall"
```

The resistance to radiation can be computed according to:

$$R_{rad} = \frac{1}{A_s\,\sigma\,\varepsilon\left(T_{ext}^2 + T_{space}^2\right)\left(T_{ext} + T_{space}\right)} \tag{4}$$

However, T_{ext} - the external surface of the space suit, is not known. Therefore, we will guess or assume this temperature and subsequently complete the problem by calculating this value and removing this assumption. A reasonable guess is T_{ext} = 250 K. An energy balance on the heater (recall that the heater power is to be selected so that $\dot{q}_{skin}$ = 0 and therefore T_{htr} = T_b) leads to:

$$\dot{q}_{out} = \frac{\left(T_b - T_{space}\right)}{R_{ins} + R_{ext} + R_{rad}} \tag{5}$$

```
T_ext=250 [K]                                          "guess for exterior wall outside temperature"
R_rad=1/(emm*A*sigma#*(T_ext^2+T_space^2)*(T_ext+T_space)) "radiation resistance"
q_dot_out=(T_b-T_space)/(R_cond_ins+R_cond_ext+R_rad)      "rate of heat transfer"
```

The problem is solved and the guess values in EES updated (select Update Guesses from the Calculate menu). The assumed value of T_{ext} is commented out and T_{ext} is recalculated according to:

$$T_{ext} = T_b - \dot{q}_{out}\left(R_{ins} + R_{ext}\right) \tag{6}$$

```
{T_ext=250 [K]}                             "guess for exterior wall outside temperature"
T_ext=T_b-q_dot_out*(R_cond_ins+R_cond_ext)    "recalculate exterior wall outside temperature"
```

which leads to $\dot{q}_{out}$ = 69 W.

b.) In order, rate the importance of the following design parameters to your result from (a): k_{ins}, k_{ext}, and ε.

The magnitude of the thermal resistances that participate in the process are R_{ins} = 0.67 K/W, R_{ext} = 6.9x10^{-5} K/W, and R_{rad} = 3.77 K/W. In a series resistance circuit, the largest resistors dominate and therefore the most important parameters are those that dictate R_{rad} and the least important are those that determine R_{ext}. In order, the most important parameters are ε and k_{ins}. The value of k_{ext} is almost completely unimportant.

c.) Plot the heat transfer per unit area required to eliminate heat loss as a function of the emissivity, ε.

A parametric table is created (select New Parametric Table from the Tables menu) and the parameters emm and q_dot_out are added (Figure 4).



Figure 4: New Parametric Table window.

The value of emm is varied from 0.1 to 1.0 by right-clicking on the column heading and selecting Alter Values (Figure 5)



Figure 5: Alter values to vary emissivity in the parametric table.

The specified value of emissivity is commented out in the program and the table is run (select Solve Table from the Calculate menu). Select New Plot (X-Y Plot) from the Plots menu to generate Figure 6.



Figure 6: Required heat transfer per unit area as a function of the suit emissivity.

While the average emissivity of the suit's external surface is $\varepsilon = 0.25$, you have found that this value can change substantially based on how dirty or polished the suit is. You are worried about these local variations causing the astronaut discomfort due to local hot and cold spots.

d.) Assume that the heater power is kept at the value calculated in (a). Plot the rate of heat transfer from the skin as a function of the fractional change in the emissivity of the suit surface.

The code for part (a) is commented out:

```
{"part (a)"
R_cond_liner=th_liner/(k_liner*A)                    "resistance to conduction through liner"
R_cond_ins=th_ins/(k_ins*A)                          "resistance to conduction through the insulation"
R_cond_ext=th_ext/(k_ext*A)                          "resistance to conduction through exterior wall"
{T_ext=250 [K]}                                      "guess for exterior wall outside temperature"
R_rad=1/(emm*A*sigma#*(T_ext^2+T_space^2)*(T_ext+T_space)) "radiation resistance"
q_dot_out=(T_b-T_space)/(R_cond_ins+R_cond_ext+R_rad)
T_ext=T_b-q_dot_out*(R_cond_ins+R_cond_ext)          "recalculate exterior wall outside temperature"}
```

and the heat transfer rate is set according to the result calculated in (a). The emissivity is varied from its nominal value by an amount *fct* - the fractional change:

$$\varepsilon_{dirty} = fct\,\varepsilon \tag{7}$$

```
"part (d)"
fct=1.5 [-]                        "fractional change in the emissivity"
emm_dirty=emm*fct                  "emissivity at a location where suit has gotten tarnished"
q_dot_htr=69 [W]                   "heat transfer rate calculated in (a)"
```

The resistance of the liner, insulation, and external layer are computed as before:

```
R_cond_liner=th_liner/(k_liner*A)          "resistance to conduction through liner"
```

```
R_cond_ins=th_ins/(k_ins*A)                "resistance to conduction through the insulation"
R_cond_ext=th_ext/(k_ext*A)                "resistance to conduction through exterior wall"
```

The external surface temperature, T_{ext}, is again assumed and the assumed value is used to compute R_{rad}:

```
T_ext=250 [K]                                        "guess for exterior wall outside temperature"
R_rad=1/(emm_dirty*A*sigma#*(T_ext^2+T_space^2)*(T_ext+T_space)) "radiation resistance"
```

The energy balance on the heater is:

$$\dot{q}_{htr} = \frac{\left(T_{htr} - T_{space}\right)}{R_{ins} + R_{ext} + R_{rad}} + \frac{\left(T_{htr} - T_b\right)}{R_{liner}} \tag{8}$$

```
q_dot_htr=(T_htr-T_space)/(R_cond_ins+R_cond_ext+R_rad)+(T_htr-T_b)/R_cond_liner
        "energy balance on heater"
```

The heat transfer rate to space is:

$$\dot{q}_{out} = \frac{\left(T_{htr} - T_{space}\right)}{R_{ins} + R_{ext} + R_{rad}} \tag{9}$$

```
q_dot_out=(T_htr-T_space)/(R_cond_ins+R_cond_ext+R_rad)    "heat transfer to space"
```

The problem is solved and the guess values in EES updated (select Update Guesses from the Calculate menu). The assumed value of T_{ext} is commented out and T_{ext} is recalculated according to:

$$T_{ext} = T_{htr} - \dot{q}_{out}\left(R_{ins} + R_{ext}\right) \tag{10}$$

```
{T_ext=250 [K]}                                "guess for exterior wall outside temperature"
T_ext=T_htr-q_dot_out*(R_cond_ins+R_cond_ext)  "recalculate exterior wall outside temperature"
```

The heat transfer rate from the skin is:

$$\dot{q}_{skin} = \frac{\left(T_b - T_{htr}\right)}{R_{liner}} \tag{11}$$

```
q_dot_skin=(T_b-T_htr)/R_cond_liner     "heat transfer from tissue"
```

which leads to $\dot{q}_{skin}$ = 15.6 W. Figure 7 illustrates the rate of heat transfer from the skin as a function of the fractional change in the emissivity of the suit surface.



Figure 7: Heat transfer from the skin as a function of the fractional change in emissivity.