Problem 1.6-2 (1-14 in text): Optimizing a Heat Sink

Your company has developed a micro-end milling process that allows you to easily fabricate an array of very small fins in order to make heat sinks for various types of electrical equipment. The end milling process removes material in order to generate the array of fins. Your initial design is the array of pin fins shown in Figure P1.6-2. You have been asked to optimize the design of the fin array for a particular application where the base temperature is T_{base} = 120°C and the air temperature is T_{air} = 20°C. The heat sink is square; the size of the heat sink is W = 10 cm. The conductivity of the material is k = 70 W/m-K. The distance between the edges of two adjacent fins is a, the diameter of a fin is D, and the length of each fin is L.



Figure P1.6-2: Pin fin array

Air is forced to flow through the heat sink by a fan. The heat transfer coefficient between the air and the surface of the fins as well as the unfinned region of the base, $\bar{h}$, has been measured for the particular fan that you plan to use and can be calculated according to:

$$\bar{h} = 40\left[\frac{\text{W}}{\text{m}^2\ \text{K}}\right]\left(\frac{a}{0.005\ [\text{m}]}\right)^{0.4}\left(\frac{D}{0.01\ [\text{m}]}\right)^{-0.3}$$

Mass is not a concern for this heat sink; you are only interested in maximizing the heat transfer rate from the heat sink to the air given the operating temperatures. Therefore, you will want to make the fins as long as possible. However, in order to use the micro-end milling process you cannot allow the fins to be longer than 10x the distance between two adjacent fins. That is, the length of the fins may be computed according to: $L = 10\,a$. You must choose the most optimal value of a and D for this application.

a.) Prepare a model using EES that can predict the heat transfer coefficient for a given value of a and D. Use this model to predict the heat transfer rate from the heat sink for a = 0.5 cm and D = 0.75 cm.

The input values are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
```

```
T_air=converttemp(C,K,20)                    "air temperature"
T_base=converttemp(C,K,120)                  "base temperature"
k=70 [W/m-K]                                 "fin material conductivity"
W=10.0 [cm]*convert(cm,m)                    "base width"
```

The optimization parameters, *a* and *D*, are set to their initial values:

```
"Optimization parameters"
a=0.5 [cm]*convert(cm,m)                     "distance between adjacent fins"
D=0.75 [cm]*convert(cm,m)                    "diameter of fins"
```

The length of the fins is computed using the aspect ratio and the number of fins is determined according to:

$$N = \left(\frac{W}{a+D}\right)^2 \tag{1}$$

The heat transfer coefficient is computed using the equation provided in the problem statement.

```
L=10*a                                                      "length of fins"
N=(W/(a+D))^2                                               "number of fins"
h = 40 [W/m^2-K]*(a/0.005 [m])^(0.4)*(D/0.01 [m])^(-0.3)    "heat transfer coefficient"
```

The perimeter and cross-sectional area of each fin are computed according to:

$$p = \pi D \tag{2}$$

$$A_c = \pi \frac{D^2}{4} \tag{3}$$

The EES function for the fin efficiency of a constant cross-sectional area fin is used. The function is accessed using the Function Information selection from the Options menu and then selecting Fin Efficiency from the pull-down menu. Scroll to the Circular-Base Rectangular Fin (Figure 2(a)) and select Info to learn how to access this function (Figure 2(b)).



(a) (b)

Figure 2: (a) Function Information window and (b) Help information for the Circular-Base Rectangular Fin.

The fin constant, *mL*, is computed according to:

$$mL = \sqrt{\frac{h\,p}{k\,A_c}}\,L \tag{4}$$

and used to call the function eta_fin_spine_rect which returns the fin efficiency, η_f.

```
p=pi*D                          "perimeter of fin"
Ac=pi*D^2/4                     "cross sectional area of fin"
mL=L*sqrt(h*p/(k*Ac))           "fin constant"
eta=eta_fin_spine_rect(mL)      "fin efficiency"
```

The total area of the fins on the heat sink is:

$$A_f = N\,p\,L \tag{5}$$

and so the total resistance of the fins are:

$$R_f = \frac{1}{h\,A_f\,\eta_f} \tag{6}$$

```
A_f=p*L*N                       "finned area"
R_f=1/(h*A_f*eta)               "resistance of fins"
```

The total area of the base of the heat sink that is not finned is:

$$A_{uf} = W^2 - N\,A_c \tag{7}$$

and the thermal resistance from the unfinned base is:

$$R_{uf} = \frac{1}{h\,A_{uf}} \tag{8}$$

```
A_uf=W^2-N*Ac                                      "unfinned area"
R_uf=1/(h*A_uf)                                    "resistance of unfinned area"
```

The total resistance of the heat sink is the combination of R_f and R_{uf} in parallel:

$$R_{total} = \left(\frac{1}{R_f} + \frac{1}{R_{uf}}\right)^{-1} \tag{9}$$

and the total heat transfer rate is:

$$\dot{q} = \frac{\left(T_{base} - T_{air}\right)}{R_{total}} \tag{10}$$

```
R_total=(1/R_f+1/R_uf)^(-1)
   "total thermal resistance of the heat sink"
q_dot=(T_base-T_air)/R_total                       "heat transfer"
```

which leads to $\dot{q}$ = 291.7 W.

b.) Prepare a plot that shows the heat transfer rate from the heat sink as a function of the distance between adjacent fins, *a*, for a fixed value of *D* = 0.75 cm. Be sure that the fin length is calculated using $L = 10\,a$. Your plot should exhibit a maximum value, indicating that there is an optimal value of *a*.

Figure 3 illustrates the heat transfer rate from the heat sink as a function of *a* for *D* = 0.75 cm.



Figure 3: Heat transfer rate as a function of the distance between adjacent fins for *D* = 0.75 cm.

c.) Prepare a plot that shows the heat transfer rate from the heat sink as a function of the diameter of the fins, *D*, for a fixed value of *a* = 0.5 cm. Be sure that the fin length is calculated using $L = 10\,a$. Your plot should exhibit a maximum value, indicating that there is an optimal value of *D*.

Figure 4 illustrates the heat transfer rate from the heat sink as a function of *D* for *a* = 0.5 cm.



Figure 4: Heat transfer rate as a function of the diameter of the fins for *a* = 0.5 cm.

d.) Determine the optimal value of *a* and *D* using EES' built-in optimization capability.

Comment out the optimization parameters (*a* and *D*) and access the optimization algorithms from the Calculate Menu by selecting Min/Max (Figure 5).



Figure 5: Find Minimum or Maximum Window

Select the variable to be minimized or maximized from the list on the left and the independent variables to be varied from the list on the right. You will need to provide a reasonable initial guess and bounds for the independent variables by selecting the Bounds button; note that it is not practical for *a* or *D* to be less than 1.0 mm. You can experiment with the different optimization methods and see which technique is more robust.

I found the genetic optimization algorithm to work the best for this problem; with a sufficient number of individuals I identified an optimal design consisting of approximately 1500 very small fins of D = 1.1 mm separated by a = 1.4 mm. The associated rate of heat transfer is $\dot{q}$ = 352.2 W.