Problem 1.9-6

Your company manufacturers heater wire. Heater wire is applied to surfaces that need to be heated and then current is passed through the wire in order to develop ohmic dissipation. A key issue with your product is failures that occur when a length of the wire becomes detached from the surface and therefore the wire is not well connected thermally to the surface. The wire in the detached region tends to get very hot and melt.

The wire diameter is D = 0.4 mm and the current passing through the wire is *current* = 10 amp. the detached wire is exposed to surroundings at T_∞ = 20ºC through convection and radiation. The average convection heat transfer coefficient is $\overline{h}$ = 30 W/m^2-K and the emissivity of the wire surface is ε = 0.5. The length of the wire that is detached from the surface is L = 1 cm. The ends of the wire at x = 0 and x = L are held at T_{end} = 50ºC. The wire conductivity is k = 10 W/m-K and the electrical resistivity is ρ_e = 1x10^{-7} ohm-m.

a.) Develop a numerical model of the wire and use the model to plot the temperature distribution within the wire.

The inputs are entered in EES and, looking ahead to parts (d) and (e), functions are defined that return the conductivity and electrical resistivity:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

function k(T)
"Input
T - temperature (K)
Output
k - conductivity (W/m-K)"
    k_o=10 [W/m-K]
    k=k_o
end

function rho_e(T)
"Input
T - temperature (K)
Output
rho_e - electrical resistivity (ohm-m)"
    rho_e_o=1e-7 [ohm-m]
    rho_e=rho_e_o
end

"Inputs"
L_cm=1 [cm]                                "length of detached region, in cm"
L=L_cm*convert(cm,m)                       "length of detached region"
d=0.4 [mm]*convert(mm,m)                   "diameter of heater wire"
current=10 [amp]                           "current"
h_bar=30 [W/m^2-K]                         "heat transfer coefficient"
e=0.5 [-]                                  "emissivity"
T_end=converttemp(C,K,50 [C])              "end temperatures"
T_infinity=converttemp(C,K,20[C])          "surrounding temperatures"
```

The nodes are positioned along the wire according to:

$$x_i = \frac{(i-1)L}{(N-1)} \quad \text{for } i = 1..N \tag{1}$$

where N is the number of nodes. The distance between adjacent nodes is:

$$\Delta x = \frac{L}{(N-1)} \tag{2}$$

The perimeter and cross-sectional area for conduction along the wire is:

$$per = \pi\, D \tag{3}$$

$$A_c = \pi \frac{D^2}{4} \tag{4}$$

```
N=21 [-]                                        "number of nodes"
duplicate i=1,N
     x[i]=L*(i-1)/(N-1)                         "position of nodes"
end
Dx=L/(N-1)                                      "distance between nodes"
per=pi*d                                        "perimeter of wire"
A_c=pi*d^2/4                                    "cross-sectional area of wire"
```

The temperatures of the nodes at the ends of the wire are set:

$$T_1 = T_{end} \tag{5}$$

$$T_N = T_{end} \tag{6}$$

```
T[1]=T_end
T[N]=T_end
```

Energy balances on the internal nodes lead to:

$$\begin{gathered} k_{T=\frac{(T_i+T_{i-1})}{2}} A_c \frac{(T_{i-1}-T_i)}{\Delta x} + k_{T=\frac{(T_i+T_{i+1})}{2}} A_c \frac{(T_{i+1}-T_i)}{\Delta x} + \rho_{e,T=T_i} \frac{\Delta x}{A_c} current^2 + \\ \varepsilon\, per\, \Delta x\, \sigma \left(T_\infty^4 - T_i^4\right) + \bar{h}\; per\, \Delta x \left(T_\infty - T_i\right) = 0 \\ i = 2..(N-1) \end{gathered} \tag{7}$$

```
duplicate i=2,(N-1)
     k((T[i]+T[i-1])/2)*A_c*(T[i-1]-T[i])/Dx+k((T[i]+T[i+1])/2)*A_c*(T[i+1]-T[i])/Dx&
     +rho_e(T[i])*Dx*current^2/A_c+e*per*Dx*sigma#*(T_infinity^4-T[i]^4)+h_bar*per*Dx*(T_infinity-T[i])=0
end
duplicate i=1,N
```

```
    T_C[i]=converttemp(K,C,T[i])                                        "temperature in C"
end
```

Figure 1 illustrates the temperature as a function of position in the wire.



Figure 1: Temperature as a function of position.

b.) Plot the maximum temperature in the wire as a function of the number of nodes in the numerical model.

The maximum temperature (T_{max}) is identified.

```
T_max_C=max(T_C[1..N])          "maximum temperature in C"
```

Figure 2 illustrates the maximum temepraturе as a function of number of nodes.



Figure 2: Maximum temperature as a function of the number of nodes.

c.) Plot the maximum temperature in the wire as a function of the length of the detached region. If the maximum temperature of the wire before failure is T_{max} = 400ºC, then what is the maximum allowable length of detached wire?

Figure 3 illustrates the maximum temperature in the wire as a function of the length of the detached region and shows that the detached region cannot exceed about 0.75 cm in length without resulting in failure.



Figure 3: Maximum temperature as a function of the detached length for the nominal wire with *k* = 10 W/m-K and ρ_e = 1x10^{-7} ohm-m as well as for materials D and E.

You are looking at methods to alleviate this problem and have identified an alternative material, material D, in which the electrical resistivity is ρ_e = 1x10^{-7} ohm-m but the conductivity increases with temperature according to:

$$k = 10\left[\frac{\text{W}}{\text{m K}}\right] + 0.05\left[\frac{\text{W}}{\text{m K}^2}\right]\left(T - 300\left[\text{K}\right]\right)$$

d.) Modify your numerical model in order to simulate material D. Overlay on your plot from (c) the maximum temperature as a function of the length of the detached wire for material D.

The function for the conductivity is modified:

```
function k(T)
"Input
T - temperature (K)
Output
k - conductivity (W/m-K)"
     k_o=10 [W/m-K]
     alpha=0.05 [W/m-K^2]
{    k=k_o}
     k=k_o+alpha*(T-300 [K])
end
```

The maximum temperature as a function of detached length is shown in Figure 3. The maximum length of detached wire has increased to approximately 1.1 cm.

You have identified another alternative material, material E, in which the thermal conductivity is k = 10 W/m-K but the electrical resistivity decreases with temperature according to:

$$\rho = 1\text{x}10^{-7}\ \left[\text{ohm m}\right] - 1\text{x}10^{-10}\ \left[\frac{\text{ohm m}}{\text{K}}\right]\left(T - 300\left[\text{K}\right]\right)$$

e.) Modify your numerical model in order to simulate material E. Overlay on your plot from (c) the maximum temperature as a function of the length of the detached wire for material E.

The function for the conductivity is restored and the function for electrical resistivity is modified:

```
function k(T)
"Input
T - temperature (K)
Output
k - conductivity (W/m-K)"
     k_o=10 [W/m-K]
     alpha=0.05 [W/m-K^2]
     k=k_o
{    k=k_o+alpha*(T-300 [K])}
end

function rho_e(T)
"Input
T - temperature (K)
Output
rho_e - electrical resistivity (ohm-m)"
     rho_e_o=1e-7 [ohm-m]
     beta=1e-10 [ohm-m/K]
     {rho_e=rho_e_o}
     rho_e=rho_e_o-beta*(T-300 [K])
end
```

The maximum temperature as a function of detached length is shown in Figure 3. The maximum length of detached wire has increased to approximately 1.0 cm.