Problem 1.2-10: Insulation Conductivity Test

You have been contracted by ASHRAE (the American Society of Heating, Refrigeration, and Air-Conditioning Engineers) to measure the thermal conductivity of various, new materials for insulating pipes. Your contract specifies that you will measure the thermal conductivity to within 10%. Your initial design for the test setup is shown in Figure P1.2-10. The test facility consists of a pipe (with conductivity k_{pipe} = 120 W/m-K) with inner diameter, $D_{i,pipe}$ = 6.0 inch and thickness th_{pipe} = 0.5 inch that carries a flow of chilled water, T_{water} = 10°C. The heat transfer coefficient between the water and the internal surface of the pipe is $\overline{h}_{water}$ = 300 W/m^2-K. The pipe is covered by a th_{ins} = 2.0 inch thick layer of the insulation (with conductivity k_{ins}) that is being tested. Two thermocouples are embedded in the insulation, one connected to the outer surface ($T_{ins,out}$) and the other to the inner surface ($T_{ins,in}$). The insulation material is surrounded by a th_m = 3.0 inch thick layer of a material with a well-known thermal conductivity, k_m = 2.0 W/m-K. Two thermocouples are embedded in the material at its inner and outer surface ($T_{m,in}$ and $T_{m,out}$, respectively). Finally, a band heater is wrapped around the outer surface of the material. Assume that the thickness of the band heater is negligibly small. The band heater provides $\dot{q}_{band}$ = 3 kW/m. The outer surface of the band heater is exposed to ambient air at T_{air} = 20°C and has a heat transfer coefficient, $\overline{h}_{air}$ = 10 W/m^2-K and emissivity ε = 0.5. A contact resistance of $R''_c = 1\text{x}10^{-4}$ m^2-K/W is present at all 3 interfaces in the problem (i.e., between the pipe and the insulation, the insulation and the material, and the material and the band heater).



Figure P1.2-10: Test facility for measuring pipe insulation

You may assume that the problem is 1-D (i.e., there are no variations along the length or circumference of the pipe) and do the problem on a per unit length of pipe (L=1 m) basis.

a.) Draw a resistance network that represents the test facility. Clearly label each resistance and indicate what it represents. Be sure to indicate where in the network the heat input from the band heater will be applied and also the location of the thermocouples mentioned in the problem statement.

The resistance network is shown in Figure 2 and includes convection with the water and the air ($R_{conv,w}$ and R_{air}), conduction through the pipe, insulation, and material (R_{pipe}, $R_{cond,ins}$, and $R_{cond,m}$), contact resistances between the pipe and insulation ($R_{c,1}$), the insulation and material ($R_{c,2}$), and the material and the band heater ($R_{c,3}$), and radiation (R_{rad}).



Figure 2: Resistance network representing the test facility.

b.) If the coefficient of performance (COP) of the refrigeration system is nominally 3.5, then how much heat must be rejected to the ambient air (W)? Recall that COP is the ratio of the amount of refrigeration provided to the amount of input power required.

The known information is entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
k_pipe=120 [W/m-K]                          "pipe conductivity"
D_i_pipe=6.0 [inch]*convert(inch,m)         "pipe inner diameter"
th_pipe=0.5 [inch]*convert(inch,m)          "pipe thickness"
T_water=converttemp(C,K,10 [C])             "water temperature"
h_water=300 [W/m^2-K]                       "water to pipe heat transfer coefficient"
th_ins=2.0 [inch]*convert(inch,m)           "insulation thickness"
k_m=2.0 [W/m-K]                             "material thermal conductivity"
th_m=3.0 [inch]*convert(inch,m)             "material thickness"
T_air=converttemp(C,K,20 [C])               "air temperature"
h_air=10 [W/m^2-K]                          "air to heater heat transfer coefficient"
e=0.5 [-]                                   "emissivity of band heater surface"
R``_c=1e-4 [m^2-K/W]                        "contact resistance"
L=1 [m]                                     "length of pipe"
q_dot_htr=3 [kW]*convert(kW,W)              "heater power"
k_ins=1.0 [W/m-K]                           "insulation conductivity"
```

The values of the convection resistances are computed:

$$R_{conv,w} = \frac{1}{h_{water}\,\pi\,L\,D_{i,pipe}} \tag{1}$$

$$R_{conv,air} = \frac{1}{h_{air}\,\pi\,L\left(D_{i,pipe} + 2\,th_{pipe} + 2\,th_{ins} + 2\,th_{m}\right)} \tag{2}$$

```
R_conv_w=1/(h_water*pi*L*D_i_pipe)                            "pipe-to-water convection"
R_conv_air=1/(h_air*pi*L*(D_i_pipe+2*th_pipe+2*th_ins+2*th_m))    "heater to air convection"
```

The conduction resistances are calculated according to:

$$R_{pipe}=\frac{\ln\left(\frac{D_{i,pipe}+2\,th_{pipe}}{D_{i,pipe}}\right)}{2\,\pi\,L\,k_{pipe}} \tag{3}$$

$$R_{cond,ins}=\frac{\ln\left(\frac{D_{i,pipe}+2\,th_{pipe}+2\,th_{ins}}{D_{i,pipe}+2\,th_{pipe}}\right)}{2\,\pi\,L\,k_{ins}} \tag{4}$$

$$R_{cond,m}=\frac{\ln\left(\frac{D_{i,pipe}+2\,th_{pipe}+2\,th_{ins}+2\,th_{m}}{D_{i,pipe}+2\,th_{pipe}+2\,th_{ins}}\right)}{2\,\pi\,L\,k_{m}} \tag{5}$$

```
R_pipe=ln((D_i_pipe/2+th_pipe)/(D_i_pipe/2))/(2*pi*L*k_pipe)       "pipe conduction resistance"
R_cond_ins=ln((D_i_pipe/2+th_pipe+th_ins)/(D_i_pipe/2+th_pipe))/(2*pi*L*k_ins)
        "insulation conduction resistance"
R_cond_m=ln((D_i_pipe/2+th_pipe+th_ins+th_m)/(D_i_pipe/2+th_pipe+th_ins))/(2*pi*L*k_m)
        "material conduction resistance"
```

The contact resistances are calculated according to:

$$R_{c,1}=\frac{R''_c}{\pi\,L\left(D_{i,pipe}+2\,th_{pipe}\right)} \tag{6}$$

$$R_{c,2}=\frac{R''_c}{\pi\,L\left(D_{i,pipe}+2\,th_{pipe}+2\,th_{ins}\right)} \tag{7}$$

$$R_{c,3}=\frac{R''_c}{\pi\,L\left(D_{i,pipe}+2\,th_{pipe}+2\,th_{ins}+2\,th_{m}\right)} \tag{8}$$

```
R_c_1=R``_c/(pi*(D_i_pipe+2*th_pipe)*L)                    "pipe-to-insulation contact resistance"
R_c_2=R``_c/(pi*(D_i_pipe+2*th_pipe+2*th_ins)*L)           "insulation-to-material contact resistance"
R_c_3=R``_c/(pi*(D_i_pipe+2*th_pipe+2*th_ins+2*th_m)*L)  "material-to-heater contact resistance"
```

Finally, the radiation resistance is calculated according to:

$$R_{rad} = \frac{1}{\pi L \left(D_{i,pipe} + 2\,th_{pipe} + 2\,th_{ins} + 2\,th_m \right) \sigma \varepsilon \left(T_{htr}^2 + T_{air}^2 \right) \left(T_{htr} + T_{air} \right)} \qquad (9)$$

but T_{htr} is not known in Eq. (9); therefore, a guess value of T_{htr} must be used to allow the calculation of the resistance. This guess value will be removed once a solution is obtained. A reasonable guess value for the heater temperature is something higher than the ambient temperature.

```
T_htr_g=500 [K]
    "this is a guess for the heater temperature - it allows me to calculate the radiation resistance"
    "this guess will be removed to complete the solution"
R_rad=1/(pi*(D_i_pipe+2*th_pipe+2*th_ins+2*th_m)*L*sigma#*e*(T_htr_g^2+T_air^2)*(T_htr_g+T_air))
                    "radiation resistance"
```

The heat transferred to the heater must either pass inwards to the water or outwards to the ambient air.

$$\dot{q}_{htr} = \frac{\left(T_{htr} - T_{water} \right)}{R_{conv,w} + R_{pipe} + R_{c,1} + R_{cond,ins} + R_{c,2} + R_{cond,m} + R_{c,3}} + \frac{\left(T_{htr} - T_{air} \right)}{\left(\frac{1}{R_{conv,air}} + \frac{1}{R_{rad}} \right)^{-1}} \qquad (10)$$

```
q_dot_htr=(T_htr-T_water)/(R_conv_w+R_pipe+R_c_1+R_cond_ins+R_c_2+R_cond_m+R_c_3)+(T_htr-
T_air)/((1/R_conv_air+1/R_rad)^(-1))      "heater power"
```

The calculated and guessed values of T_{htr} will not be the same (unless you are very lucky); update the guess values for the calculation (select Update Guesses from the Calculate menu) and then specify that T_htr_g must equal T_htr. You will need to comment the assignment of T_htr_g to avoid over-specifying the problem.

```
{T_htr_g=500 [K]                                              "this is a guess for the heater
temperature - it allows me to calculate the radiation resistance"
        "this guess will be removed to complete the solution"}
T_htr=T_htr_g
```

The heater temperature is found to be 394.6 K (about 120°C which is too hot to touch). The heat transferred through the insulation is:

$$\dot{q}_{ins} = \frac{\left(T_{htr} - T_{water} \right)}{R_{conv,w} + R_{pipe} + R_{c,1} + R_{cond,ins} + R_{c,2} + R_{cond,m} + R_{c,3}} \qquad (11)$$

```
q_dot_ins=(T_htr-T_water)/(R_conv_w+R_pipe+R_c_1+R_cond_ins+R_c_2+R_cond_m+R_c_3)
                  "heat transferred through insulation"
```

which leads to $\dot{q}_{ins}$ = 976.3 W (most of the heat is transferred to the surrounding air).

When the test facility is operated, the heater power is not measured nor are the contact resistances, emissivity, or heat transfer coefficients known with any precision. Also, the insulation thermal conductivity is not known, but rather must be calculated from the measured temperatures. The heat transferred through the material with known thermal conductivity is the same as the heat transferred through the insulation that is being measured. Therefore:

$$\dot{q}_{ins} = \frac{T_{m,out} - T_{m,in}}{R_{cond,m}} = \frac{T_{ins,out} - T_{ins,in}}{R_{cond,ins}} \tag{12}$$

and so the resistance of the insulation can be calculated based on the ratio of the temperature differences:

$$R_{cond,ins} = \frac{\left(T_{ins,out} - T_{ins,in}\right)}{\left(T_{m,out} - T_{m,in}\right)} R_{cond,m} \tag{13}$$

Equation (13) indicates that the test facility relies on accurately measuring the temperature differences across the insulation and the temperature difference across the material.

c.) Using your model, predict the temperature difference across the insulation ($\Delta T_{ins} = T_{ins,out} - T_{ins,in}$) and the material ($\Delta T_m = T_{m,out} - T_{m,in}$).

Using Eq. (12), the two temperature differences are:

$$\Delta T_{ins} = \dot{q}_{ins}\, R_{cond,ins} \tag{14}$$

$$\Delta T_m = \dot{q}_{ins}\, R_{cond,m} \tag{15}$$

```
DeltaT_ins=q_dot_ins*R_cond_ins                     "insulation temperature difference"
DeltaT_m=q_dot_ins*R_cond_m                         "material temperature difference"
```

which leads to ΔT_{ins}= 70.2 K and ΔT_m = 33.8 K.

d.) Prepare a plot showing ΔT_m as a function of the material thickness (th_m) for thicknesses ranging from 5.0 mm to 50 cm. Explain the shape of your plot.

The plot requested by the problem statement is generated using a parametric table that include the variables th_m and DeltaT_m. The result is shown in Figure 3.



Figure 3: Temperature difference across the material as a function of the material thickness.

Notice that at low th_m the value of ΔT_m is small because the resistance of the material is small. At high values of th_m the resistance of the material is large but the heat transferred through the material becomes small (more of the energy is transferred to the air) and so the value of ΔT_m begins to decrease.

e.) Based on your plot from part (d), what is a reasonable value for th_m? Remember that you need to measure the temperature difference and therefore you would like it to be large.

A value of th_m around 10 cm provides a large value of ΔT_m; further increases are probably not justified. A similar plot and design point could be obtained for ΔT_{ins} by varying th_{ins}.