Problems 1.2-12 (1-5 in text): Floor Heater

You have decided to install a strip heater under the linoleum in your bathroom in order to keep your feet warm on cold winter mornings. Figure P1.2-12 illustrates a cross-section of the bathroom floor. The bathroom is located on the first story of your house and is W = 2.5 m wide by L = 2.5 m long. The linoleum thickness is th_L = 5 mm and has conductivity k_L = 0.05 W/m-K. The strip heater under the linoleum is negligibly thin. Beneath the heater is a piece of plywood with thickness th_p = 5 mm and conductivity k_p = 0.4 W/m-K. The plywood is supported by th_s = 6 cm thick studs that are W_s = 4 cm wide with thermal conductivity k_s = 0.4 W/m-K. The center-to-center distance between studs is p_s = 25.0 cm. Between each stud are pockets of air that can be considered to be stagnant with conductivity k_{air} = 0.025 W/m-K. A sheet of drywall is nailed to the bottom of the studs. The thickness of the drywall is th_d = 9.0 mm and the conductivity of drywall is k_d = 0.1 W/m-K. The air above in the bathroom is at $T_{air,1}$ = 15°C while the air in the basement is at $T_{air,2}$ = 5°C. The heat transfer coefficient on both sides of the floor is $\bar{h}$ = 15 W/m^2-K. You may neglect radiation and contact resistance for this problem.



Figure P1.2-12: Bathroom floor with heater.

a.) Draw a thermal resistance network that can be used to represent this situation. Be sure to label the temperatures of the air above and below the floor ($T_{air,1}$ and $T_{air,2}$), the temperature at the surface of the linoleum (T_L), the temperature of the strip heater (T_h), and the heat input to the strip heater ($\dot{q}_h$) on your diagram.

The resistance diagram corresponding to this problem is shown in Figure 2.



Figure 2: Resistance diagram representing the bathroom floor.

Starting at the left-hand side of Figure 2 (i.e., from the basement air), the resistances correspond to convection between the air in the basement and the drywall (R_{conv}), conduction through the drywall (R_d), conduction through the air (R_{air}) and studs (R_s) in parallel, conduction through the plywood (R_p), conduction through the linoleum (R_L), and convection between the air in the bathroom and the linoleum (R_{conv}).

b.) Compute the value of each of the resistances from part (a).

The known values for the problem are entered in EES and converted to base SI units:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
W=2.5 [m]                                   "width of bathroom"
L=2.5 [m]                                   "length of bathroom"
T_air_1=converttemp(C,K,15)                 "air temperature in the bathroom"
T_air_2=converttemp(C,K,5)                  "air temperature in the basement"
h=15 [W/m^2-K]                              "heat transfer coefficient"
th_L=5.0 [mm]*convert(mm,m)                 "linoleum thickness"
k_L=0.05 [W/m-K]                            "linoleum thermal conductivity"
th_P=5.0 [mm]*convert(mm,m)                 "plywood thickness"
k_P=0.4 [W/m-K]                             "plywood thermal conductivity"
th_s=6.0 [cm]*convert(cm,m)                 "stud thickness"
W_s=4.0 [cm]*convert(cm,m)                  "stud width"
k_s=0.4 [W/m-K]                             "stud conductivity"
p_s=25 [cm]*convert(cm,m)                   "stud center-to-center distance"
k_air=0.025 [W/m-K]                         "air conductivity"
th_d=9.0 [mm]*convert(mm,m)                 "drywall thickness"
k_d=0.1 [W/m-K]                             "drywall conductivity"
```

The units for each of these variables is set in the Variable Information window (select Variable Information from the Options menu), Figure 3. The units of each of the additional variables that are added as you solve the problem should be set in the Variable Information window.



Variable	Guess	Lower	Upper	Display			Units	Key	Comment
h	15	-infinity	infinity	A	3	N	W/m^2-K		
k_air	0.025	-infinity	infinity	A	3	N	W/m-K		
k_d	0.1	-infinity	infinity	A	3	N	W/m-K		
k_L	0.05	-infinity	infinity	A	3	N	W/m-K		
k_P	0.4	-infinity	infinity	A	3	N	W/m-K		
k_s	0.4	-infinity	infinity	A	3	N	W/m-K		
L	2.5	-infinity	infinity	A	3	N	m		
p_s	0.25	-infinity	infinity	A	3	N	m		
th_d	0.009	-infinity	infinity	A	3	N	m		
th_L	0.005	-infinity	infinity	A	3	N	m		
th_P	0.005	-infinity	infinity	A	3	N	m		
th_s	0.06	-infinity	infinity	A	3	N	m		
T_air_1	288.2	-infinity	infinity	A	3	N	K		

Figure 3: Variable Information window.

The area of the floor is:

$$A = L\,W \tag{1}$$

The convection resistance is computed according to:

$$R_{conv} = \frac{1}{h\,A} \tag{2}$$

The conduction resistances of the linoleum, plywood, and drywall are computed:

$$R_L = \frac{th_L}{k_L\,A} \tag{3}$$

$$R_p = \frac{th_p}{k_p\,A} \tag{4}$$

$$R_d = \frac{th_d}{k_d\,A} \tag{5}$$

```
A=L*W                          "area for conduction through floor"
R_conv=1/(h*A)                 "convection resistance"
R_L=th_L/(k_L*A)               "linoleum resistance"
R_P=th_P/(k_P*A)               "plywood resistance"
R_d=th_d/(k_d*A)               "drywall resistance"
```

The conduction resistance of the studs is computed according to:

$$R_s = \frac{th_s}{k_s\, A\left(\frac{W_s}{p_s}\right)} \tag{6}$$

Note that the area for conduction is the product of the area of the floor and the fraction of the floor occupied by the studs. The conduction resistance of the air is:

$$R_{air} = \frac{th_s}{k_{air}\, A\left(\frac{p_s - W_s}{p_s}\right)} \tag{7}$$

```
R_s=th_s/(k_s*A*W_s/p_s)                    "stud resistance"
R_air=th_s/(k_air*A*(p_s-W_s)/p_s)          "air resistance"
```

These calculations lead to R_{conv} = 0.011 K/W, R_L = 0.016 K/W, R_p = 0.002 K/W, R_d = 0.014 K/W, R_s = 0.15 K/W, and R_{air} = 0.46 K/W.

c.) How much heat must be added by the heater to raise the temperature of the floor to a comfortable 20°C?

If T_s in Figure 2 is 20°C then the heat transferred to the bathroom ($\dot{q}_1$) is:

$$\dot{q}_1 = \frac{T_L - T_{air,1}}{R_{conv}} \tag{8}$$

```
T_L=converttemp(C,K,20)                     "linoleum temperature"
q_dot_1=(T_L-T_air_1)/R_conv                "heat transferred to bathroom"
```

which leads to $\dot{q}_1$ = 469 W. The temperature of the heater is therefore:

$$T_h = T_L + \dot{q}_1\, R_L \tag{9}$$

```
T_h=T_L+q_dot_1*R_L                         "strip heater temperature"
```

which leads to a heater temperature, T_h = 300.7 K. The heater must provide $\dot{q}_2$ (the heat transferred to the bathroom) as well as $\dot{q}_1$ (the heat transferred to the basement).

$$\dot{q}_1 = \frac{T_h - T_{air,2}}{R_{conv} + R_d + \left[\frac{1}{R_s} + \frac{1}{R_{air}}\right]^{-1} + R_p} \tag{10}$$

and the total heater power is:

$$\dot{q}_h = \dot{q}_1 + \dot{q}_2 \tag{11}$$

```
q_dot_2=(T_h-T_air_2)/(R_conv+R_d+(1/R_s+1/R_air)^(-1)+R_P) "heat lost to lower story"
q_dot_h=q_dot_1+q_dot_2                                     "total heater power"
```

which leads to $\dot{q}_2$ = 161 W and $\dot{q}_h$ = 630 W.

d.) What physical quantities are most important to your analysis? What physical quantities are unimportant to your analysis?

Figure 4 illustrates the values of the resistances on the resistance diagram.



Figure 4: Resistance values.

Examination of Figure 4 shows that R_p, R_d, and R_{conv} are unimportant relative to the amount of heat transferred to the basement; these resistances are small in a series combination. Therefore, you can expect that the conductivity of the drywall and plywood as well as their thickness are not very important. Furthermore, the resistance of the air is larger than the resistance of the studs; in a parallel combination, the larger resistance is not important. Therefore, the conductivity of air is likely not very important. The important quantities include the conductivity of the studs and their size as well as the thickness and conductivity of the linoleum and its thickness. The heat transfer coefficient is also important.

e.) Discuss at least one technique that could be used to substantially reduce the amount of heater power required while still maintaining the floor at 20°C. Note that you have no control over $T_{air,1}$ or h.

The heat transferred to the bathroom is given by Eq. (8); you cannot change h and therefore the value of R_{conv} is fixed. The only way for you can reduce the heater power is to reduce the amount of heat transferred to the basement. This can be done most effectively by increasing the resistance of the studs, perhaps by increasing their thickness or reducing their width.