Problem 1.4-5

Solve the problem stated in EXAMPLE 1.3-2 numerically rather than analytically.

a.) Develop a numerical model that can predict the temperature distribution within the lens. Prepare a plot of the temperature as a function of position.

The inputs are entered in EES:

```
"Problem 1.4-5"
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
q``_rad=0.1 [W/cm^2]*convert(W/cm^2,W/m^2)          "radiation incident on the lens"
L=1.0 [cm]*convert(cm,m)                            "thickness of lens"
T_a=converttemp(C,K,20)                             "ambient temperature"
h=20 [W/m^2-K]                                      "heat transfer coefficient"
k=1.5 [W/m-K]                                       "conductivity of lens"
alpha=0.1 [1/mm]*convert(1/mm,1/m)                  "absorption coefficient"
A=1 [m^2]                                           "per unit area"
```

Nodes are distributed uniformly throughout the computational domain; the distance between adjacent nodes is:

$$\Delta x = \frac{L}{(N-1)} \tag{1}$$

where N is the number of nodes. The position of each node is:

$$x_i = \Delta x\,(i-1) \quad \text{for } i = 1...N \tag{2}$$

```
N=10 [-]                                            "number of nodes"
Dx=L/(N-1)                                          "distance between adjacent nodes"
duplicate i=1,N
    x[i]=Dx*(i-1)                                   "position of each node"
end
```

An energy balance on an internal control volume is shown in Figure 1.



Figure 1: Energy balance on an internal control volume

The energy balance is:

$$\dot{q}_{top} + \dot{q}_{bottom} + \dot{g} = 0 \tag{3}$$

Substituting rate equations into Eq. (3) leads to:

$$\frac{k\,A}{\Delta x}\left(T_{i-1} - T_i\right) + \frac{k\,A}{\Delta x}\left(T_{i+1} - T_i\right) + A\,\Delta x\,\dot{q}''_{rad}\,\alpha\exp\left(-\alpha\,x_i\right) = 0 \tag{4}$$

```
"internal control volume energy balances"
duplicate i=2,(N-1)
    k*A*(T[i-1]-T[i])/Dx+k*A*(T[i+1]-T[i])/Dx+A*Dx*q``_rad*alpha*exp(-alpha*x[i])=0
end
```

An energy balance on node 1 located at the upper surface is shown in Figure 2.



Figure 2: Energy balance on the upper edge control volume

The energy balance for node 1 is:

$$\dot{q}_{conv} + \dot{q}_{bottom} + \dot{g} = 0 \tag{5}$$

Substituting rate equations into Eq. (5) leads to:

$$h\,A\left(T_a - T_1\right) + \frac{k\,A}{\Delta x}\left(T_2 - T_1\right) + \frac{A\,\Delta x}{2}\,\dot{q}''_{rad}\,\alpha\exp\left(-\alpha\,x_1\right) = 0 \tag{6}$$

The corresponding energy balance for node *N* located at the lower surface is:

$$h\,A\left(T_a - T_N\right) + \frac{k\,A}{\Delta x}\left(T_{N-1} - T_N\right) + \frac{A\,\Delta x}{2}\,\dot{q}''_{rad}\,\alpha\exp\left(-\alpha\,x_N\right) = 0 \tag{7}$$

```
"upper edge"
h*A*(T_a-T[1])+k*A*(T[2]-T[1])/Dx+A*Dx*q``_rad*alpha*exp(-alpha*x[1])/2=0

"lower edge"
h*A*(T_a-T[N])+k*A*(T[N-1]-T[N])/Dx+A*Dx*q``_rad*alpha*exp(-alpha*x[N])/2=0
```

The temperature distribution in the lens is shown in Figure 3.



Figure 3: Temperature as a function of position in the lens

b.) Plot some characteristic of your solution as a function of the number of nodes to show that you are using a sufficient number of nodes.

The maximum temperature in the lens is obtained using the `Max` command in EES:

```
T_max=MAX(T[1..N])                                    "maximum temperature in lens"
```

The maximum temperature and number of nodes are placed in a parametric table; the number of nodes is varied and the results are shown in Figure 4.



Figure 4: Maximum temperature as a function of the number of nodes

c.) Think of a sanity check that you can use to gain confidence in your model; that is, can you change some input parameter and show that the solution behaves as you would expect. Support your answer with a plot.

As the lens conductivity becomes very large, the temperature rise within the lens should be reduced. Figure 3 illustrates the predicted result when the conductivity is increased by a factor of 10, to 15 W/m-K.

d.) Plot the maximum lens temperature as a function of the heat transfer coefficient, $\overline{h}$.

Figure 5 illustrates the maximum temperature in the lens as a function of the heat transfer coefficient.



Figure 5: Maximum temperature as a function of the heat transfer coefficient