Problem 1.7-4 (1-16 in text): Solar Collector Tube

The receiver tube of a concentrating solar collector is shown in Figure P1.7-4.



Figure P1.7-4: A solar collector

The receiver tube is exposed to solar radiation that has been reflected from a concentrating mirror. The heat flux received by the tube is related to the position of the sun and the geometry and efficiency of the concentrating mirrors. For this problem, you may assume that all of the radiation heat flux is absorbed by the collector and neglect the radiation emitted by the collector to its surroundings. (Chapter 10 will provide information on the radiation characteristics of surfaces that will allow a more complete evaluation of solar collectors.) The flux received at the collector surface ($\dot{q}''_s$) is not circumferentially uniform but rather varies with angular position; the flux is uniform along the top of the collector, $\pi < \phi < 2\pi$ rad, and varies sinusoidally along the bottom, $0 < \phi < \pi$ rad, with a peak at $\phi = \pi/2$ rad.

$$\dot{q}''_s(\phi) = \begin{cases} \dot{q}''_t + \left(\dot{q}''_p - \dot{q}''_t\right)\sin(\phi) & \text{for } 0 < \phi < \pi \\ \dot{q}''_t \text{ for } \pi < \phi < 2\pi \end{cases}$$

where $\dot{q}''_t$ = 1000 W/m^2 is the uniform heat flux along the top of the collector tube and $\dot{q}''_p$ = 5000 W/m^2 is the peak heat flux along the bottom. The receiver tube has an inner radius of r = 5.0 cm and thickness of th = 2.5 mm (because $th/r \ll 1$ it is possible to ignore the small difference in convection area on the inner and outer surfaces of the tube). The thermal conductivity of the tube material is k = 10 W/m-K. The solar collector is used to heat water, which is at T_w = 80°C at the axial position of interest. The average heat transfer coefficient between the water and the internal surface of the collector is $\overline{h}_w$ = 100 W/m^2-K. The external surface of the collector is exposed to air at T_a = 25°C. The average heat transfer coefficient between the air and the external surface of the collector is $\overline{h}_a$ = 25 W/m^2-K.

a.) Can the collector be treated as an extended surface for this problem (i.e., can the temperature gradients in the radial direction in the collector material be neglected)?

The input parameters are entered into the EES program:

```
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
qf_t=1000 [W/m^2]                        "Heat flux on top surface"
qf_p=5000 [W/m^2]                        "Peak heat flux on bottom surface"
k = 10 [W/m-K]                           "Conductivity of the collector material"
th=2.5 [mm]*convert(mm,m)                "thickness of collector"
r=5.0[cm]*convert(cm,m)                  "inner radius of collector"
T_a=converttemp(C,K,25[C])               "temperature of surrounding air"
h_bar_a=25 [W/m^2-K]                     "heat transfer coefficient to surrounding air"
T_w=converttemp(C,K,80 [C])              "temperature of water"
h_bar_w=100 [W/m^2-K]                    "heat transfer coefficient to water"
L=1 [m]                                  "length of the collector"
```

The extended surface approximation neglects temperature gradients across the thickness of the tube but considers the temperature difference between the collector surfaces and the surrounding water and air. This assumption is equivalent to neglecting the conduction resistance across the tube as being small relative to the two convection resistances characterizing heat transfer to the air and the water. Two Biot numbers are calculated based on the air (Bi_a) and water (Bi_w)

$$Bi_a = \frac{\overline{h}_a\, th}{2\, k}$$

$$Bi_w = \frac{\overline{h}_w\, th}{2\, k}$$

```
Bi_a=h_bar_a*th/(2*k)                    "Biot number based on air side"
Bi_w=h_bar_w*th/(2*k)                    "Biot number based on water side"
```

Both Biot numbers are found to be much less than one (0.003 and 0.01, respectively) and therefore the extended surface approximation is valid.

b.) Develop an analytical model that will allow the temperature distribution in the collector wall to be determined as a function of circumferential position.

The computational domain (the receiver tube wall) goes from $\phi = 0$ to 2π rad; however, there are actually two separate computational domains, the top and bottom, each with a different governing equation due to the different spatial variation in the heat flux. A differential control volume is used to derive the governing equation in the bottom computational domain. An energy balance on the control volume leads to:

$$\dot{q}_\phi + \dot{q}_s = \dot{q}_{\phi+d\phi} + \dot{q}_{conv,a} + \dot{q}_{conv,w}$$

The conduction per unit length in the circumferential direction, $\dot{q}_\phi$, is written using Fourier's law.

$$\dot{q}_{\phi} = -k\,\frac{th\,L}{r}\frac{dT_b}{d\phi}$$

where L is the length of the collector tube and T_b is the temperature along the bottom of the tube. Note that the temperature gradient along the circumference of the tube (K/m) is written as the product the derivative of temperature with respect to angle (K/rad) and the inverse of the radius.

The convection to the air and the water are:

$$\dot{q}_{conv,a} = r\,d\phi\,L\,\overline{h}_a\left(T_b - T_a\right)$$

$$\dot{q}_{conv,w} = r\,d\phi\,L\,\overline{h}_w\left(T_b - T_w\right)$$

The energy absorbed due to the solar flux is:

$$\dot{q}_s = \left[\dot{q}_t'' + \left(\dot{q}_p'' - \dot{q}_t''\right)\sin\left(\phi\right)\right]L\,r\,d\phi$$

Combining these equations leads to:

$$\left[\dot{q}_t'' + \left(\dot{q}_p'' - \dot{q}_t''\right)\sin\left(\phi\right)\right]r\,d\phi = \frac{d}{d\phi}\left[-k\,\frac{th}{r}\frac{dT_b}{d\phi}\right]d\phi + r\,d\phi\,\overline{h}_a\left(T_b - T_a\right) + r\,d\phi\,\overline{h}_w\left(T_b - T_w\right)$$

which can be simplified to:

$$\frac{d^2T_b}{d\phi^2} - \frac{r^2\,\overline{h}_a}{k\,th}T_b - \frac{r^2\,\overline{h}_w}{k\,th}T_b = -\frac{r^2\,\overline{h}_a}{k\,th}T_a - \frac{r^2\,\overline{h}_w}{k\,th}T_w - \left[\dot{q}_t'' + \left(\dot{q}_p'' - \dot{q}_t''\right)\sin\left(\phi\right)\right]\frac{r^2}{k\,th} \tag{1}$$

Equation (1) is the governing differential equation for the temperature along bottom of the collector and it is therefore valid from $0 < \theta < \pi$. A similar set of steps leads to the governing equation for the temperature along the top of the collector (T_t) that is valid from $\pi < \theta < 2\,\pi$.

$$\frac{d^2T_t}{d\phi^2} - \frac{r^2\,\overline{h}_a}{k\,th}T_t - \frac{r^2\,\overline{h}_w}{k\,th}T_t = -\frac{r^2\,\overline{h}_a}{k\,th}T_a - \frac{r^2\,\overline{h}_w}{k\,th}T_w - \dot{q}_t''\frac{r^2}{k\,th} \tag{2}$$

Equations (1) and (2) are entered into Maple and solved in order to determine their general solutions.

```
> restart;
> ODE_b:=diff(diff(T_b(phi),phi),phi)-r^2*h_bar_a*T_b(phi)/(k*th)-r^2*h_bar_w*T_b(phi)/(k*th)=-
r^2*h_bar_a*T_a/(k*th)-r^2*h_bar_w*T_w/(k*th)-(qf_t+(qf_p-qf_t)*sin(phi))*r^2/(k*th);
```

$$ODE\_b := \left(\frac{d^2}{d\phi^2}\mathrm{T\_b}(\phi)\right) - \frac{r^2\, h\_bar\_a\, \mathrm{T\_b}(\phi)}{k\, th} - \frac{r^2\, h\_bar\_w\, \mathrm{T\_b}(\phi)}{k\, th} = -\frac{r^2\, h\_bar\_a\, T\_a}{k\, th} - \frac{r^2\, h\_bar\_w\, T\_w}{k\, th} - \frac{(qf\_t + (qf\_p - qf\_t)\sin(\phi))\, r^2}{k\, th}$$

\> T_b_s:=dsolve(ODE_b);

$$T\_b\_s := \mathrm{T\_b}(\phi) = \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\phi}{\sqrt{k}\sqrt{th}}\right)}\_C2 + \mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\phi}{\sqrt{k}\sqrt{th}}\right)}\_C1 + (-r^2(-qf\_p + qf\_t)(h\_bar\_a + h\_bar\_w)\sin(\phi) + (r^2(h\_bar\_a + h\_bar\_w) + k\, th)(h\_bar\_a\, T\_a + h\_bar\_w\, T\_w + qf\_t)) \big/ ((r^2(h\_bar\_a + h\_bar\_w) + k\, th)(h\_bar\_a + h\_bar\_w))$$

\> ODE_t:=diff(diff(T_t(phi),phi),phi)-r^2*h_bar_a*T_t(phi)/(k*th)-r^2*h_bar_w*T_t(phi)/(k*th)=-r^2*h_bar_a*T_a/(k*th)-r^2*h_bar_w*T_w/(k*th)-qf_t*r^2/(k*th);

$$ODE\_t := \left(\frac{d^2}{d\phi^2}\mathrm{T\_t}(\phi)\right) - \frac{r^2\, h\_bar\_a\, \mathrm{T\_t}(\phi)}{k\, th} - \frac{r^2\, h\_bar\_w\, \mathrm{T\_t}(\phi)}{k\, th} = -\frac{r^2\, h\_bar\_a\, T\_a}{k\, th} - \frac{r^2\, h\_bar\_w\, T\_w}{k\, th} - \frac{qf\_t\, r^2}{k\, th}$$

\> T_t_s:=dsolve(ODE_t);

$$T\_t\_s := \mathrm{T\_t}(\phi) = \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\phi}{\sqrt{k}\sqrt{th}}\right)}\_C2 + \mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\phi}{\sqrt{k}\sqrt{th}}\right)}\_C1 + \frac{h\_bar\_a\, T\_a + h\_bar\_w\, T\_w + qf\_t}{h\_bar\_a + h\_bar\_w}$$

Note that although both solutions are given with constants of integration C_1 and C_2 it is clear that these constants cannot be the same. Here, the constants for the general solution for T_t will be C_3 and C_4. The solutions are copied into EES and manipulated slightly to obtain:

```
"Solutions"
T_b = exp(r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*phi)*C_2+&
    exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*phi)*C_1+&
    (-r^2*(h_bar_a+h_bar_w)*(-qf_p+qf_t)*sin(phi)+(h_bar_a*T_a+&
    T_w*h_bar_w+qf_t)*(r^2*(h_bar_a+h_bar_w)+k*th))/(h_bar_a+&
    h_bar_w)/(r^2*(h_bar_a+h_bar_w)+k*th)
T_t = exp(r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*phi)*C_4+&
    exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*phi)*C_3+&
        (h_bar_a*T_a+h_bar_w*T_w+qf_t)/(h_bar_a+h_bar_w)
```

There are four unknown constants of integration (C_1 through C_4). Therefore, 4 boundary conditions are required to obtain the four constants of integration (two for each solution). The temperature must be continuous at both of the interfaces between the top and bottom domains:

$$T_{b,\phi=0} = T_{t,\phi=2\pi} \tag{3}$$

$$T_{b,\phi=\pi} = T_{t,\phi=\pi} \qquad (4)$$

Also, the energy flowing between the regions must be conserved. An interface energy balance at $\phi = \pi$ rad provides:

$$-k \frac{L\,th}{r} \frac{dT_b}{d\phi}\bigg|_{\phi=\pi} = -k \frac{L\,th}{r} \frac{dT_t}{d\phi}\bigg|_{\phi=\pi}$$

which implies that the temperature gradient at $\phi = \pi$ rad is continuous in both domains:

$$\frac{dT_b}{d\phi}\bigg|_{\phi=\pi} = \frac{dT_t}{d\phi}\bigg|_{\phi=\pi} \qquad (5)$$

A similar equation results for the interface at $\phi = 0$ rad.

$$\frac{dT_b}{d\phi}\bigg|_{\phi=0} = \frac{dT_t}{d\phi}\bigg|_{\phi=2\pi} \qquad (6)$$

Maple can carry out the symbolic manipulation of the solution while EES does the algebra to determine the constants. To obtain the left and right hand sides of Eq. (3):

```
> T_b_0:=rhs(eval(T_b_s,phi=0));
```

$$T\_b\_0 := \_C2 + \_C1 + \frac{h\_bar\_a\ T\_a + h\_bar\_w\ T\_w + qf\_t}{h\_bar\_a + h\_bar\_w}$$

```
> T_t_2pi:=rhs(eval(T_t_s,phi=2*pi));
```

$$T\_t\_2pi := \mathbf{e}^{\left(\frac{2\, r \sqrt{h\_bar\_a + h\_bar\_w}\ \pi}{\sqrt{k}\ \sqrt{th}}\right)} \_C2 + \mathbf{e}^{\left(-\frac{2\, r \sqrt{h\_bar\_a + h\_bar\_w}\ \pi}{\sqrt{k}\ \sqrt{th}}\right)} \_C1 + \frac{h\_bar\_a\ T\_a + h\_bar\_w\ T\_w + qf\_t}{h\_bar\_a + h\_bar\_w}$$

These two expressions can be cut and pasted into EES and, with minimal modification, used to set the boundary condition associated with Eq. (3). The necessary modifications include changing $\_C_1$ and _C2 to C_1 and C_2 in the equation for T_b_0 and changing _C1 and _C2 to C_3 and C_4 in the equation for T_t_2pi.

```
"Boundary conditions"
"Temperature equality at phi=0"
T_b_0 = C_2+ C_1+(h_bar_a*T_a+h_bar_w*T_w+qf_t)/(h_bar_a+h_bar_w)
    "temperature in bottom domain at phi=0"
T_t_2pi = exp(2*r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_4+&
    exp(-2*r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_3+&
    (h_bar_a*T_a+h_bar_w*T_w+qf_t)/(h_bar_a+h_bar_w)
    "temperature in top domain at phi=2 pi"
```

```
T_b_0=T_t_2pi
```

The EES and Maple text listed above seems long and complicated however very little of it needed to be entered manually; the process of solving a relatively complex heat transfer problem is reduced to a relatively straightforward integration of two powerful pieces of software.

The process is repeated for Eq. (4), in Maple:

```
> T_b_pi:=rhs(eval(T_b_s,phi=pi));
```

$$T\_b\_pi := \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a+h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C2+\mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a+h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C1+\big(-r^2\,(-qf\_p+qf\_t)\,(h\_bar\_a+h\_bar\_w)\sin(\pi)+(r^2\,(h\_bar\_a+h\_bar\_w)+k\,th)\,(h\_bar\_a\;T\_a+h\_bar\_w\;T\_w+qf\_t)\big)\Big/\big((r^2\,(h\_bar\_a+h\_bar\_w)+k\,th)\,(h\_bar\_a+h\_bar\_w)\big)$$

```
> T_t_pi:=rhs(eval(T_t_s,phi=pi));
```

$$T\_t\_pi := \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a+h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C2+\mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a+h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C1+\frac{h\_bar\_a\;T\_a+h\_bar\_w\;T\_w+qf\_t}{h\_bar\_a+h\_bar\_w}$$

The symbolic equations determined by Maple are then entered in EES with the same modification for _C1 and _C2 noted above:

```
"Temperature equality at pi"
T_b_pi = exp(r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_2+&
     exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_1+&
     (-r^2*(-qf_p+qf_t)*(h_bar_a+h_bar_w)*sin(pi)+(r^2*(h_bar_a+h_bar_w)+k*th)&
     *(h_bar_a*T_a+h_bar_w*T_w+qf_t))/(r^2*(h_bar_a+h_bar_w)+k*th)/(h_bar_a+h_bar_w)
     "temperature in bottom domain at phi = pi"
T_t_pi = exp(r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_4+&
     exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_3+&
     (h_bar_a*T_a+h_bar_w*T_w+qf_t)/(h_bar_a+h_bar_w)
     "temperature in top domain at phi = pi"
T_b_pi=T_t_pi
```

Equations (5) and (6) are dealt with in the same way. Maple is used to evaluate the symbolic expressions for the required derivatives:

```
> dTbdphi_0:=rhs(eval(diff(T_b_s,phi),phi=0));
```

$$dTbdphi\_0 := \frac{r\sqrt{h\_bar\_a+h\_bar\_w}\;\_C2}{\sqrt{k}\sqrt{th}}-\frac{r\sqrt{h\_bar\_a+h\_bar\_w}\;\_C1}{\sqrt{k}\sqrt{th}}-\frac{r^2\,(-qf\_p+qf\_t)}{r^2\,(h\_bar\_a+h\_bar\_w)+k\,th}$$

```
> dTtdphi_2pi:=rhs(eval(diff(T_t_s,phi),phi=2*pi));
```

$$dTtdphi\_2pi := \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(\frac{2\, r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C2}{\sqrt{k}\sqrt{th}} - \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(-\frac{2\, r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C1}{\sqrt{k}\sqrt{th}}$$

> dTbdphi_pi:=rhs(eval(diff(T_b_s,phi),phi=pi));

$$dTbdphi\_pi := \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C2}{\sqrt{k}\sqrt{th}} - \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C1}{\sqrt{k}\sqrt{th}} - \frac{r^2\,(-qf\_p + qf\_t)\cos(\pi)}{r^2\,(h\_bar\_a + h\_bar\_w) + k\, th}$$

> dTtdphi_pi:=rhs(eval(diff(T_t_s,phi),phi=pi));

$$dTtdphi\_pi := \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C2}{\sqrt{k}\sqrt{th}} - \frac{r\sqrt{h\_bar\_a + h\_bar\_w}\, \mathbf{e}^{\left(-\frac{r\sqrt{h\_bar\_a + h\_bar\_w}\,\pi}{\sqrt{k}\sqrt{th}}\right)}\_C1}{\sqrt{k}\sqrt{th}}$$

These expressions are entered in EES (with changes to _C1 and _C2) in order to provide the final two boundary conditions.

```
"Temperature gradient equality at 0"
dTbdphi_0 = r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*C_2-&
    r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*C_1-&
    r^2*(-qf_p+qf_t)/(r^2*(h_bar_a+h_bar_w)+k*th)
    "gradient in bottom domain at phi =0"
dTtdphi_2pi = r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*exp(2*r/k^(1/2)/th^(1/2)*&
    (h_bar_a+h_bar_w)^(1/2)*pi)*C_4-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*&
    exp(-2*r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_3
    "gradient in top at phi = 2 pi"
dTbdphi_0=dTtdphi_2pi

"Temperature gradient equality at pi"
dTbdphi_pi = r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*exp(r/k^(1/2)/th^(1/2)*&
    (h_bar_a+h_bar_w)^(1/2)*pi)*C_2-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*&
    exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_1-r^2*(-qf_p+qf_t)*cos(pi)&
    /(r^2*(h_bar_a+h_bar_w)+k*th)
    "gradient in bottom at phi = pi"
dTtdphi_pi = r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*exp(r/k^(1/2)/th^(1/2)*&
```

```
    (h_bar_a+h_bar_w)^(1/2)*pi)*C_4-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)&
    *exp(-r/k^(1/2)/th^(1/2)*(h_bar_a+h_bar_w)^(1/2)*pi)*C_3
    "gradient in top at phi = pi"
dTbdphi_pi=dTtdphi_pi
```

The analytical solution is converted to Celsius:

```
T_b_C=converttemp(K,C,T_b)                              "bottom temperature in C"
T_t_C=converttemp(K,C,T_t)                              "top temperature in C"
```

and plotted by setting up two parametric tables. The first table contains the variables phi and T_b_C (where phi is varied from 0 to π) while the second table contains the variables phi and T_t_C (where phi is varied from π to 2π). The temperature distribution is shown in Figure 2.



Figure 2: Temperature distribution around the circumference of the collector tube for various values of the tube conductivity.

It is possible to adjust any of the input parameters within EES the solution remains valid because the constants are evaluated symbolically. For example, Figure 2 also illustrates how the solution varies as the conductivity of the receiver tube changes.