Problem 1.2-23
Figure P1.2-23 illustrates a cross-sectional view of a water heater.



Figure P1.2-23: Water heater.

The water flows through a tube with inner diameter D_{in}= 0.75 inch and outer diameter D_{out} = 0.875 inch. The conductivity of the tube material is k_{tube} = 12 W/m-K. The water in the tube is at mean temperature T_f = 50°C and pressure p_f = 18 psi. The heat transfer coefficient between the water and the internal surface of the tube is $\overline{h}_{in}$ = 250 W/m^2-K. A very thin heater is wrapped around the outer surface of the tube. The heater provides a heat transfer rate of $\dot{q}''$ = 10,000 W/m^2. Insulation is wrapped around the heater. The thickness of the insulation is th_{ins} = 0.5 inch and the conductivity is k_{ins}= 0.5 W/m-K. There is a contact resistance between the heater and the tube and the heater and the insulation. The area specific contact resistance for both interfaces is R''_c = 1x10^{-4} K-m^2/W. The outer surface of the insulation radiates and convects to surroundings at T_∞ = 20°C. The heat transfer coefficient between the surface of the insulation and the air is $\overline{h}_{out}$ = 15 W/m^2-K. The emissivity of the outer surface of the insulation is ε = 0.5.

a.) Draw a resistance network that represents this problem. Label each resistance and clearly indicate what it represents. Show where the heater power enters your network.

The resistance network is shown in Figure 2.



The resistances include:
$R_{conv,in}$ = convection to water
$R_{cond,tube}$ = conduction through tube
R_c = contact resistance
$R_{cond,ins}$ = conduction through insulation
$R_{conv,out}$ = convection resistance to air
R_{rad} = radiation resistance

Figure 2: Resistance network.

b.) Using EES, determine the temperature of the heater and the rate of heat transfer to the water.

The inputs are entered in EES; note that the problem is done on a per unit length basis, L= 1 m.

```
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
D_in=0.75 [inch]*convert(inch,m)                "inner diameter"
D_out=0.875 [inch]*convert(inch,m)              "outer diameter"
k_tube=12 [W/m-K]                               "conductivity of tube"
T_f=converttemp(C,K,50 [C])                     "fluid temperature"
p_f=18 [psi]*convert(psi,Pa)                    "fluid pressure"
T_infinity=converttemp(C,K,20 [C])              "ambient temperature"
h_bar_out=15 [W/m^2-K]                          "heat transfer coefficient to ambient air"
h_bar_in=250 [W/m^2-K]                          "heat transfer coefficient to fluid"
e=0.5 [-]                                       "emissivity of surface of insulation"
k_ins=0.5 [W/m-K]                               "conductivity of insulation"
th_ins_inch=0.5 [inch]                          "thickness of insulation, in inch"
th_ins=th_ins_inch*convert(inch,m)              "thickness of insulation"
R``=1e-4 [K-m^2/W]                              "area specific contact resistance"
q``=10000 [W/m^2]                               "heat flux provided by heater"
L=1 [m]                                         "per unit length basis"
```

The conduction resistance of the tube and the insulation is:

$$R_{cond,tube} = \frac{\ln\left(\frac{D_{out}}{D_{in}}\right)}{2\,\pi\,k_{tube}\,L} \tag{1}$$

$$R_{cond,ins} = \frac{\ln\left(\dfrac{D_{out} + 2\,th_{ins}}{D_{out}}\right)}{2\,\pi\,k_{ins}\,L} \tag{2}$$

The contact resistance is:

$$R_c = \frac{R''}{\pi\,D_{out}\,L} \tag{3}$$

The convection resistance between the water and the tube surface is:

$$R_{conv,in} = \frac{1}{\bar{h}_{in}\,\pi\,D_{in}\,L} \tag{4}$$

$$R_{conv,out} = \frac{1}{\bar{h}_{out}\,\pi\left(D_{out} + 2\,th_{ins}\right)L} \tag{5}$$

```
R_cond_tube=ln(D_out/D_in)/(2*pi*k_tube*L)                "tube conduction resistance"
R_cond_ins=ln((D_out+2*th_ins)/D_out)/(2*pi*k_ins*L)      "insulation conduction resistance"
R_c=R``/(pi*D_out*L)                                      "contact resistance"
R_conv_in=1/(h_bar_in*pi*D_in*L)                          "internal convection resistance"
R_conv_out=1/(h_bar_out*pi*(D_out+2*th_ins)*L)            "external convection resistance"
```

In order to calculate the radiation resistance, the external surface temperature T_{ext}, is assumed. The radiation resistance is:

$$R_{rad} = \frac{1}{\varepsilon\,\sigma\,\pi\left(D_{out} + 2\,th_{ins}\right)L\left(T_{ext}^2 + T_\infty^2\right)\left(T_{ext} + T_\infty\right)} \tag{6}$$

```
T_ext=T_infinity                              "guess for external surface temperature"
R_rad=1/(e*sigma#*pi*(D_out+2*th_ins)*L*(T_ext^2+T_infinity^2)*(T_ext+T_infinity))
                                              "radiation resistance"
```

An energy balance on the heater is:

$$\dot{q}''\,\pi\,D_{out} = \frac{\left(T_h - T_\infty\right)}{R_c + R_{cond,ins} + \left[\dfrac{1}{R_{rad}} + \dfrac{1}{R_{conv,out}}\right]^{-1}} + \frac{\left(T_h - T_f\right)}{R_c + R_{cond,tube} + R_{conv,in}} \tag{7}$$

```
q``*pi*D_out*L=(T_h-T_infinity)/(R_c+R_cond_ins+(1/R_rad+1/R_conv_out)^(-1))+&
        (T_h-T_f)/(R_c+R_cond_tube+R_conv_in)          "energy balance on heater"
```

The problem is solved and the guess values are updated. The assumed value of T_{ext} is commented out and then recalculated based on the solution. The rate of heat transfer to the ambient is:

$$\dot{q}_{out} = \frac{\left(T_h - T_\infty\right)}{R_c + R_{cond,ins} + \left[\dfrac{1}{R_{rad}} + \dfrac{1}{R_{conv,out}}\right]^{-1}} \tag{8}$$

and the external surface temperature is:

$$T_{ext} = T_h - \dot{q}_{out}\left(R_c + R_{cond,ins}\right) \tag{9}$$

```
{T_ext=T_infinity}                                       "guess for external surface temperature"
q_dot_out=(T_h-T_infinity)/(R_c+R_cond_ins+(1/R_rad+1/R_conv_out)^(-1))
                                                         "rate of heat transfer to ambient"
T_ext=T_h-q_dot_out*(R_c+R_cond_ins)                     "recalculate external temperature"
T_h_C=converttemp(K,C,T_h)                               "heater temperature, in C"
```

which leads to T_h = 90.85°C. The rate of heat transfer to the water is computed:

$$\dot{q}_{in} = \frac{\left(T_h - T_f\right)}{R_c + R_{cond,tube} + R_{conv,in}} \tag{10}$$

```
q_dot_in=(T_h-T_f)/(R_c+R_cond_tube+R_conv_in)     "rate of heat transfer to fluid"
```

which leads to $\dot{q}_{in}$ = 581 W.

c.) What is the efficiency of the heater (the ratio of the power provided to the water to the power provided to the heater)?

The efficiency is defined as:

$$\eta = \frac{\dot{q}_{in}}{\dot{q}_{in} + \dot{q}_{out}} \tag{11}$$

```
eta=q_dot_in/(q_dot_out+q_dot_in)     "efficiency of heater"
```

which leads to η = 83.2%.

d.) The efficiency of the heater is less than 100% due to heat lost to the atmosphere. Rank the following parameters in terms of their relative importance with respect to limiting heat loss to the atmosphere: ε, R''_c, k_{ins}, $\overline{h}_{out}$. Justify your answers using your resistance network and a discussion of the magnitude of the relevant resistances.

The resistances separating the heater from the ambient include: R_{rad} = 1.89 K/W, R_c = 0.0014 K/W, $R_{cond,ins}$ = 0.24 K/W, and $R_{conv,out}$ = 0.45 K/W. This suggests that convection is more important than radiation and also more important than either conduction or contact resistance. Conduction is the next-most important resistance followed by radiation and finally contact resistance (which is absolutely unimportant). Therefore, the relative importance is: $\overline{h}_{out}$, k_{ins}, ε, and R''_c.

e.) Plot the efficiency as a function of the insulation thickness for 0 inch < th_{ins} < 1.5 inch. Explain the shape of your plot.

Figure 3 illustrates the efficiency as a function of the insulation thickness. Notice that initially as the thickness increases the efficiency actually drops; this is because the convection resistance (which, from (d), is the most important resistance) will decrease with insulation thickness since the area for convection increases. Eventually, the insulation conduction resistance increases to the point where it becomes important and the efficiency begins to increase.



Figure 3: Efficiency as a function of the insulation thickness.

f.) The temperature on the internal surface of the tube must remain below the saturation temperature of the water in order to prevent any local boiling of the water. Based on this criteria, determine the maximum possible heat flux that can be applied to the heater (for th_{ins} = 0.5 inch).

The internal surface temperature of the tube is computed:

$$T_{s,in} = T_h - \dot{q}_{in}\left(R_c + R_{cond,tube}\right) \tag{12}$$

The saturation temperature of the water (T_{sat}) is obtained from EES' internal thermodynamic property routines. The prescribed value of the heat flux is cancelled out and, instead, $T_{s,in}$ is set equal to T_{sat}.

```
{q``=10000 [W/m^2]}                                "heat flux provided by heater"
T_s_in=T_h-q_dot_in*(R_c+R_cond_tube)              "inner surface temperature of tube"
T_sat=temperature(Water,p=p_f,x=1 [-])             "saturation temperature"
T_s_in=T_sat                                       "maximum allowable temperature"
```

which leads to $\dot{q}'' = 14{,}061$ W/m^2.