**Problem 1.2-13: Burner**

An electric burner for a stove is formed by taking a cylindrical piece of metal that is $D$ = 0.32 inch in diameter and $L$ = 36 inch long and winding it into a spiral shape. The burner consumes electrical power at a rate of $\dot{q}$ = 900 W. The burner surface has an emissivity of $\varepsilon$ = 0.80. The heat transfer coefficient between the burner and the surrounding air ($\bar{h}$) depends on the surface temperature of the burner ($T_s$) according to:

$$\bar{h} = 10.7\left[\frac{\text{W}}{\text{m}^2\text{-K}}\right] + 0.0048\left[\frac{\text{W}}{\text{m}^2\text{-K}^2}\right]T_s \qquad (1)$$

where $\bar{h}$ is in [W/m$^2$-K] and $T_s$ is in [K]. The surroundings and the surrounding air temperature are at $T_{sur}$ = 20°C.

a.) Determine the steady state surface temperature of the burner.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
D=0.32 [inch]*convert(inch,m)                  "diameter of burner"
L=36 [inch]*convert(inch,m)                    "length of burner"
q_dot=900 [W]                                  "burner power consumption"
e=0.80 [-]                                     "burner emissivity"
T_sur=converttemp(C,K,20)                      "surrounding temperature"
```

In order to move logically through the problem solution it is best to initially assume a surface temperature, calculate the heat transfer rates, and finally adjust the surface temperature until the heat transfer rates are consistent with the problem statement. Therefore, an initial and reasonable guess for the surface temperature is made and used to compute the heat transfer coefficient with Eq. (1).

```
T_s=1000 [K]
    "initial guess for the surface temperature"
h=10.7 [W/m^2-K]+0.0048 [W/m^2-K^2]*T_s        "heat transfer coefficient"
```

The surface area of the burner is:

$$A_s = \pi\, D\, L \qquad (2)$$

The rate of convective heat transfer from the burner to the air is:

$$\dot{q}_{conv} = h\, A_s\left(T_s - T_{sur}\right) \qquad (3)$$

and the rate of radiative heat transfer from the burner to the surroundings is:

$$\dot{q}_{rad} = \sigma\,\varepsilon\,A_s\left(T_s - T_{sur}\right) \tag{4}$$

where $\sigma$ is Stefan-Boltzmann's constant.

```
A_s=pi*D*L                                          "surface area"
q_dot_conv=A_s*h*(T_s-T_sur)                        "convection"
q_dot_rad=A_s*e*sigma#*(T_s^4-T_sur^4)              "radiation"
```

At this point, it is necessary to adjust the surface temperature assumption so that the sum of the radiative and convective heat transfer rates are equal to $\dot{q}$, the energy provided to the burner. This could be done manually, adjust the variable T_s up or down as necessary. However, EES allows you to automate this process by solving the nonlinear set of equations; like any equation solver, EES begins this process from an initial set of values (guess values) for each of the variables and iteratively solves the equations over and over to minimize the error (the residual). The advantage of the solution approach provided above is that you have a "good" starting point defined (a good set of guess values); therefore, select Update Guesses from the Calculate menu in order to lock in these guess values. Then, remove the initially assumed value of the variable T_s (just comment it out – that is, highlight the line and right click on it, select Comment {} frorm the menu that will appear) and specify instead that the heat transfer rates must sum to $\dot{q}$.

$$\dot{q} = \dot{q}_{rad} + \dot{q}_{conv} \tag{5}$$

```
{T_s=1000 [K]
        "initial guess for the surface temperature"}
q_dot=q_dot_conv+q_dot_rad                          "energy balance on the burner"
```

Solving the EES program will lead to $T_s$ = 900 K.

b.) Prepare a plot showing the surface temperature as a function of the burner input power.

This is done most easily using a Parametric Table. Comment out the specified value of the burner input power:

```
{q_dot=900 [W]                                      "burner power consumption"}
```

and prepare a Parametric Table that includes the variables q_dot and T_s. Vary the value of q_dot, for example from 250 W to 1500 W, and solve the table. Plot the results to obtain Figure 2.




**Figure 2: Burner surface temperature as a function of burner power**