Problem 1.9-4 (1-20 in text)

An expensive power electronics module normally receives only a moderate current. However, under certain conditions it is possible that it might experience currents in excess of 100 amps. The module cannot survive such a high current and therefore, you have been asked to design a fuse that will protect the module by limiting the current that it can experience, as shown in Figure P1.9-4.



Figure 1.9-4: A fuse that protects a power electronics module from high current.

The space available for the fuse allows a wire that is L = 2.5 cm long to be placed between the module and the surrounding structure. The surface of the fuse wire is exposed to air at T_∞ = 20°C and the heat transfer coefficient between the surface of the fuse and the air is $\bar{h}$ = 5.0 W/m^2-K. The fuse surface has an emissivity of ε = 0.90. The fuse is made of an aluminum alloy with conductivity k = 150 W/m-K. The electrical resistivity of the aluminum alloy is ρ_e = 1x10^{-7} ohm-m and the alloy melts at approximately 500°C. Assume that the properties of the alloy do not depend on temperature. The ends of the fuse (i.e., at x=0 and x=L) are maintained at T_{end} =20°C by contact with the surrounding structure and the module. The current passing through the fuse, I, results in a uniform volumetric generation within the fuse material. If the fuse operates properly, then it will melt (i.e., at some location within the fuse, the temperature will exceed 500°C) when the current reaches 100 amp. Your job will be to select the fuse diameter; to get your model started you may assume a diameter of D = 0.9 mm. Assume that the volumetric rate of thermal energy generation due to ohmic dissipation is uniform throughout the fuse volume.

a.) Prepare a numerical model of the fuse that can predict the steady state temperature distribution within the fuse material. Plot the temperature as a function of position within the wire when the current is 100 amp and the diameter is 0.9 mm.

The input parameters are entered in EES and the volumetric generation rate is computed:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
L=2.5 [cm]*convert(cm,m)                "length"
d=0.9 [mm]*convert(mm,m)                "diameter"
T_a=converttemp(C,K,20)                 "air temperature"
T_end=converttemp(C,K,20)               "end temperature"
h=5.0 [W/m^2-K]                         "heat transfer coefficient"
e=0.90 [-]                              "emissivity"
```

```
k=150 [W/m-K]                              "conductivity"
er=1e-7 [ohm-m]                            "electrical resistivity"
T_melt=converttemp(C,K,500)                "melting temperature"
current=100 [amp]                          "current"

"Volumetric generation"
Ac=pi*d^2/4                                "cross-sectional area"
Rst=er*L/Ac                                "resistance"
w_dot_ohmic=current^2*Rst                  "total dissipation"
g```_dot=w_dot_ohmic/(Ac*L)                "volumetric rate of generation"
```

The appropriate Biot number for this case is:

$$Bi = \frac{h\,d}{2\,k} \tag{1}$$

The Biot number is calculated according to:

```
"Extended surface approximation"
Bi=h*d/(2*k)
```

The Biot number calculated by EES is much less than 1.0 and therefore the extended surface approximation is justified.

The development of the numerical model follows the same steps that were previously discussed in the context of numerical models of 1-D geometries. Nodes (i.e., locations where the temperature will be determined) are positioned uniformly along the length of the rod. The location of each node (x_i) is:

$$x_i = \frac{(i-1)}{(N-1)} L \quad i = 1..N \tag{2}$$

where N is the number of nodes used for the simulation. The distance between adjacent nodes (Δx) is:

$$\Delta x = \frac{L}{(N-1)} \tag{3}$$

This distribution is entered in EES:

```
"Setup nodes"
N=10 [-]                                   "number of nodes"
duplicate i=1,N
        x[i]=(i-1)*L/(N-1)                 "position of nodes"
end
Dx=L/(N-1)                                 "distance between nodes"
```

A control volume is defined around each node; the control surface bisects the distance between the nodes. The control volume shown in Fig. 2 is subject to conduction heat transfer at each edge ($\dot{q}_{top}$ and $\dot{q}_{bottom}$), convection ($\dot{q}_{conv}$), radiation ($\dot{q}_{rad}$), and generation ($\dot{g}$). The energy balance is:

$$\dot{q}_{top} + \dot{q}_{bottom} + \dot{q}_{conv} + \dot{q}_{rad} + \dot{g} = 0 \tag{4}$$

The conduction terms are approximated as:

$$\dot{q}_{top} = \frac{k\,\pi\,d^2}{4\,\Delta x}\left(T_{i-1} - T_i\right) \tag{5}$$

$$\dot{q}_{bottom} = \frac{k\,\pi\,d^2}{4\,\Delta x}\left(T_{i+1} - T_i\right) \tag{6}$$

The convection term is modeled according to:

$$\dot{q}_{conv} = h\,\pi\,d\,\Delta x\left(T_a - T_i\right) \tag{7}$$

The radiation term is:

$$\dot{q}_{rad} = \varepsilon\,\sigma\,\pi\,d\,\Delta x\left(T_a^4 - T_i^4\right) \tag{8}$$

The generation term is:

$$\dot{g} = \dot{g}'''\,\pi\,\frac{d^2}{4}\,\Delta x \tag{9}$$

Substituting Eqs. (5) through (9) into Eq. (4) leads to:

$$\begin{gathered}\frac{k\,\pi\,d^2}{4\,\Delta x}\left(T_{i-1} - T_i\right) + \frac{k\,\pi\,d^2}{4\,\Delta x}\left(T_{i+1} - T_i\right) + h_a\,\pi\,d\,\Delta x\left(T_a - T_i\right) \\ +\varepsilon\,\sigma\,\pi\,d\,\Delta x\left(T_a^4 - T_i^4\right) + \dot{g}'''\,\pi\,\frac{d^2}{4}\,\Delta x = 0 \quad \text{for } i = 2..\left(N-1\right)\end{gathered} \tag{10}$$

The nodes at the edges of the domain must be treated separately; the temperature at both edges of the fuse are specified:

$$T_1 = T_{end} \tag{11}$$

$$T_N = T_{end} \tag{12}$$

Equations (10) through (12) are a system of N equations in an equal number of unknown temperatures which are entered in EES:

```
"Numerical solution"
T[1]=T_end
T[N]=T_end
duplicate i=2,(N-1)
	k*pi*d^2*(T[i-1]-T[i])/(4*dx)+k*pi*d^2*(T[i+1]-T[i])/(4*dx)+pi*d*dx*h*(T_a-T[i])+pi*d*dx*e*sigma#*(T_a^4-T[i]^4)+g```_dot*pi*d^2*dx/4=0
end
duplicate i=1,N
	T_C[i]=converttemp(K,C,T[i])
end
```

Figure 2 illustrates the temperature distribution in the fuse for N = 100 nodes.



Figure 2: Temperature distribution in the fuse.

b.) Verify that your model has numerically converged by plotting the maximum temperature in the wire as a function of the number of nodes in your model.

With any numerical simulation it is important to verify that a sufficient number of nodes have been used so that the numerical solution has converged. The key result of the solution is the maximum temperature in the wire, which can be obtained using the MAX command:

```
T_max_C=max(T_C[1..N])
```

Figure 3 illustrates the maximum temperature as a function of the number of nodes and shows that the solution has converged for N greater than 100 nodes.



Figure 3: Maximum temperature as a function of the number of nodes.

c.) Prepare a plot of the maximum temperature in the wire as a function of the diameter of the wire for I=100 amp. Use your plot to select an appropriate fuse diameter.

The number of nodes was set to 100 and the plot shown in Figure 4 was generated:



Figure 4: Maximum temperature as a function of diameter.

The maximum temperature reaches 500°C when the diameter is approximately 1.15 mm; this would provide a fuse that correctly limited the current.