Problem 1.7-1: Furnace Manipulator Arm

You are designing a manipulator for use within a furnace. The arm must penetrate the side of the furnace, as shown in Figure P1.7-1. The arm has a diameter of D = 0.8 cm and protrudes L_i = 0.5 m into the furnace, terminating in the actuator that can be assumed to be adiabatic. The portion of the arm in the manipulator is exposed to flame and hot gas; these effects can be represented by a heat flux of $\dot{q}'' = 1\text{x}10^4$ W/m^2 and convection to gas at T_f = 500°C with heat transfer coefficient $\overline{h}_f$ = 50 W/m^2-K. The conductivity of the arm material is k = 150 W/m-K. The arm outside of the furnace has the same diameter and conductivity, but is exposed to air at T_a = 20°C with heat transfer coefficient $\overline{h}_a$ = 30 W/m^2-K. The length of the arm outside of the furnace is L_o = 0.75 m and this portion of the arm terminates in the motor system which can also be considered to be adiabatic.



Figure P1.7-1: Manipulator arm for a furnace.

a.) Is an extended surface model appropriate for this problem? Justify your answer.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
qf_rad=1e4 [W/m^2]                              "radiant heat flux on arm"
T_f=converttemp(C,K,500)                        "air temperature within furnace"
h_f=50 [W/m^2-K]
    "heat transfer coefficient within furnace"
L_i=0.5 [m]                                     "manipulator arm length within furnace"
D=0.8 [cm]*convert(cm,m)                        "diameter of arm"
k=150 [W/m-K]                                   "arm conductivity"
T_a=converttemp(C,K,20)                         "air temperature outside of furnace"
h_a=30 [W/m^2-K]
    "heat transfer coefficient outside of furnace"
L_o=0.75 [m]
    "manipulator arm length outside of furnace"
```

The Biot number based on the heat transfer coefficient within the furnace will be largest because the highest heat transfer coefficient exists within the furnace:

$$Bi_i = \frac{D\,h_f}{2\,k} \tag{1}$$

```
Bi_i=(D/2)*h_f/k                          "Biot number based on internal heat transfer coefficient"
```

which leads to Bi_i = 0.013 which is sufficiently less than unity to allow an extended surface approximation model to be used.

b.) Develop an analytical model of the manipulator arm; implement your model in EES. Plot the temperature as a function of axial position x (see Figure P1.7-1) for $-L_o < x < L_i$.

The differential equation that governs the temperature within the furnace (T_i) is derived using the differential energy balance shown in Figure 2.



Figure 2: Differential energy balance on manipulator arm within furnace.

The energy balance suggested by Figure 2 is:

$$\dot{q}_x + \dot{q}''_{rad}\,\pi\,D\,dx = \dot{q}_x + \frac{d\dot{q}}{dx}dx + h_f\,\pi\,D\left(T_i - T_f\right)dx \tag{2}$$

Substituting Fourier's law:

$$\dot{q}_x = -k\frac{\pi\,D^2}{4}\frac{dT_i}{dx} \tag{3}$$

into Eq. (2) leads to:

$$\dot{q}''_{rad}\,\pi\,D\,dx = \frac{d}{dx}\left[-k\frac{\pi\,D^2}{4}\frac{dT_i}{dx}\right]dx + h_f\,\pi\,D\left(T_i - T_f\right)dx \tag{4}$$

or

$$\frac{d^2T_i}{dx^2}-\frac{4\,h_f}{k\,D}T_i=-\frac{4\,h_f}{k\,D}T_f-\frac{4\,\dot{q}''_{rad}}{k\,D} \tag{5}$$

The solution to the ordinary differential equation is divided into its homogeneous (u_i) and particular (v_i) parts:

$$T_i=u_i+v_i \tag{6}$$

and substituted into Eq. (5) in order to obtain:

$$\underbrace{\frac{d^2u_i}{dx^2}-\frac{4\,h_f}{k\,D}u_i}_{\text{homogeneous equation}}+\underbrace{\frac{d^2v_i}{dx^2}-\frac{4\,h_f}{k\,D}v_i=-\frac{4\,h_f}{k\,D}T_f-\frac{4\,\dot{q}''_{rad}}{k\,D}}_{\text{particular equation}} \tag{7}$$

The solution to the homogeneous equation:

$$\frac{d^2u_i}{dx^2}-\frac{4\,h_f}{k\,D}u_i=0 \tag{8}$$

is

$$u_i=C_1\exp\left(m_i\,x\right)+C_2\exp\left(-m_i\,x\right) \tag{9}$$

where

$$m_i=\sqrt{\frac{4\,h_f}{k\,D}} \tag{10}$$

The solution to the particular equation:

$$\frac{d^2v_i}{dx^2}-\frac{4\,h_f}{k\,D}v_i=-\frac{4\,h_f}{k\,D}T_f-\frac{4\,\dot{q}''_{rad}}{k\,D} \tag{11}$$

is

$$v_i=T_f+\frac{\dot{q}''_{rad}}{h_f} \tag{12}$$

So the general solution for T_i is:

$$T_i = C_1 \exp\left(m_i\, x\right) + C_2 \exp\left(-m_i\, x\right) + T_f + \frac{\dot{q}''_{rad}}{h_f} \tag{13}$$

A similar set of steps leads to the general solution for the temperature outside of the furnace, T_o, which is valid for $-L_o < x < 0$:

$$T_o = C_3 \exp\left(m_o\, x\right) + C_4 \exp\left(-m_o\, x\right) + T_a \tag{14}$$

where

$$m_o = \sqrt{\frac{4\, h_a}{k\, D}} \tag{15}$$

The fin constants, m_i and m_o, are computed:

```
m_i=sqrt(4*h_f/(k*D))                          "fin constant inside furnace"
m_o=sqrt(4*h_a/(k*D))                          "fin constant outside of furnace"
```

The boundary conditions for the solution must be obtained at the edges of each of the computational domains (i.e., at $x= -L_o$, $x=0$, and $x=L_i$).

The two ends of the arm are adiabatic; therefore:

$$\left.\frac{dT_i}{dx}\right|_{x=L_i} = 0 \tag{16}$$

$$\left.\frac{dT_o}{dx}\right|_{x=-L_o} = 0 \tag{17}$$

The temperature at the interface between the two computational domains must be continuous:

$$T_{i,x=0} = T_{o,x=0} \tag{18}$$

Also, the rate of energy transferred from the furnace to the ambient air must be the same regardless of which side of the interface between the computational domains you are on; that is, an interface balance at $x = 0$ between the two computational domains leads to:

$$\left.\frac{dT_i}{dx}\right|_{x=0} = \left.\frac{dT_o}{dx}\right|_{x=0} \tag{19}$$

Substituting Eqs. (13) and (14) into Eqs. (16) through (19) leads to 4 equations for the constants of integration C_1 and C_4:

$$C_1\, m_i \exp\left(m_i\, L_i\right) - C_2\, m_i \exp\left(-m_i\, L_i\right) = 0 \tag{20}$$

$$C_3\, m_o \exp\left(-m_o\, L_o\right) - C_4\, m_o \exp\left(m_o\, L_o\right) \tag{21}$$

$$C_1 + C_2 + T_f + \frac{\dot{q}''_{rad}}{h_f} = C_3 + C_4 + T_a \tag{22}$$

$$C_1\, m_i - C_2\, m_i = C_3\, m_o - C_4\, m_o \tag{23}$$

These are entered in EES:

```
C_1*m_i*exp(m_i*L_i)-C_2*m_i*exp(-m_i*L_i)=0            "adiabatic end at L_i"
C_3*m_o*exp(-m_o*L_o)-C_4*m_o*exp(m_o*L_o)=0            "adiabatic end at -L_o"
C_1+C_2+T_f+qf_rad/h_f=C_3+C_4+T_a                      "continuity of temperatures at x=0"
C_1*m_i-C_2*m_i=C_3*m_o-C_4*m_o                          "energy balance at x=0"
```

The solutions are entered; note that the variable x_bar is varied from 0 to 1 within a parametric table which corresponds to x_i going from 0 to L_i and x_o going from 0 to -L_o.

```
x_i=x_bar*L_i
x_o=-x_bar*L_o
T_i=C_1*exp(m_i*x_i)+C_2*exp(-m_i*x_i)+T_f+qf_rad/h_f
T_o=C_3*exp(m_o*x_o)+C_4*exp(-m_o*x_o)+T_a
```

The temperature distribution in the arm is shown in Figure 3.



Figure 3: Temperature distribution.

c.) Prepare a plot showing the maximum temperature at the end of the arm (within the furnace) as a function of the internal length of the arm (L_i) for various values of the diameter (D).



Figure 4: Maximum temperature as a function of the arm length within the furnace for various values of the diameter.