Problem 1.2-15 (1-6 in text): The super ice-auger

You are a fan of ice fishing but don't enjoy the process of augering out your fishing hole in the ice. Therefore, you want to build a device, the super ice-auger, that melts a hole in the ice. The device is shown in Figure P1.2-15.



Figure P1.2-15: The super ice-auger.

A heater is attached to the back of a D = 10 inch plate and electrically activated by your truck battery, which is capable of providing V = 12 V and I = 150 A. The plate is th_p = 0.75 inch thick and has conductivity k_p = 10 W/m-K. The back of the heater is insulated; the thickness of the insulation is th_{ins} = 0.5 inch and the insulation has conductivity k_{ins} = 2.2 W/m-K. The surface of the insulation experiences convection with surrounding air at T_∞ = 5°C and radiation with surroundings also at T_∞ = 5°C. The emissivity of the surface of the insulation is ε = 0.9 and the heat transfer coefficient between the surface and the air is $\bar{h}$ = 50 W/m^2-K. The super ice-auger is placed on the ice and activated, resulting in a heat transfer to the plate-ice interface that causes the ice to melt. Assume that the water under the ice is at T_{ice} = 0°C so that no heat is conducted away from the plate-ice interface; all of the energy transferred to the plate-ice interface goes into melting the ice. The thickness of the ice is th_{ice} = 5 inch and the ice has density ρ_{ice} = 920 kg/m^3. The latent heat of fusion for the ice is Δi_{fus} = 333.6 kJ/kg.

a.) Determine the heat transfer rate to the plate-ice interface.

The inputs are entered in EES:

```
"P1.2-15"
$UnitSystem SI MASS  RAD PA K J
$Tabstops 0.2 0.4 0.6 0.8 3.5

D=10 [inch]*convert(inch,m)                       "diameter of ice fishing hole"
th_ins=0.5 [inch]*convert(inch,m)                 "insulation thickness"
k_ins=2.2 [W/m-K]                                 "insulation conductivity"
th_p=0.75 [inch]*convert(inch,m)                  "plate thickness"
k_p=10 [W/m-K]                                    "conductivity of plate"
e=0.9 [-]                                         "emissivity"
h_bar=50 [W/m^2-K]                                "air heat transfer coefficient"
T_infinity=converttemp(C,K,5 [C])                 "ambient temperature"
V=12 [V]                                          "battery voltage"
I=150 [A]                                         "current"
```

```
T_ice=converttemp(C,K,0 [C])                    "temperature of ice-water interface"
th_ice=5 [inch]*convert(inch,m)                 "thickness of ice"
rho_ice=920 [kg/m^3]                            "density of ice"
DELTAi_fus=333.6 [kJ/kg]*convert(kJ/kg,J/kg)    "enthalpy of fusion of ice"
```

The power provided to the heater is the product of the voltage and current:

$$\dot{q} = V\, I \tag{1}$$

```
q_dot=V*I                                       "power to melting plate"
```

A resistance network that can be used to represent this problem is shown in Figure P1.2-15-2.



The resistances include:
$R_{cond,p}$ = conduction through plate
$R_{cond,ins}$ = conduction through insulation
R_{rad} = radiation resistance
R_{conv} = convection resistance

Figure P1.2-15-2: The super ice-auger.

In order to compute the radiation resistance required to solve the problem, it is necessary to assume a value of T_{top}, the temperature at the top of the insulation (this value will eventually be commented out in order to complete the problem):

```
T_top=360 [K]            "guess for top surface temperature"
```

The cross-sectional area of the plate is computed:

$$A_c = \frac{\pi D^2}{4} \tag{2}$$

and the radiation resistance is computed according to:

$$R_{rad} = \frac{1}{\varepsilon\, A_c\, \sigma \left(T_{top}^2 + T_\infty^2\right)\left(T_{top} + T_\infty\right)} \tag{3}$$

```
A_c=pi*D^2/4                                                  "area of hole"
R_rad=1/(e*A_c*sigma#*(T_top^2+T_infinity^2)*(T_top+T_infinity))   "radiation resistance"
```

Note that the equations should be entered, the units set, and the EES code solved line by line in order to debug the code in small segments.

The convection resistance is computed according to:

$$R_{conv} = \frac{1}{A_c\,\bar{h}} \tag{4}$$

and the conduction resistances are computed according to:

$$R_{cond,ins} = \frac{th_{ins}}{A_c\,k_{ins}} \tag{5}$$

$$R_{cond,p} = \frac{th_p}{A_c\,k_p} \tag{6}$$

```
R_conv=1/(A_c*h_bar)                          "air convection resistance"
R_cond_ins=th_ins/(k_ins*A_c)                 "conduction resistance of insulation"
R_cond_p=th_p/(k_p*A_c)                       "conduction resistance of plate"
```

The heat transfer from the heater to the ambient surroundings ($\dot{q}_1$ in Figure P1.2-15-2) is:

$$\dot{q}_1 = \frac{\left(T_h - T_\infty\right)}{R_{cond,ins} + \left(\frac{1}{R_{rad}} + \frac{1}{R_{conv}}\right)^{-1}} \tag{7}$$

and the heat transfer to the ice is:

$$\dot{q}_2 = \frac{\left(T_h - T_{ice}\right)}{R_{cond,p}} \tag{8}$$

where T_h is the heater temperature. An energy balance on the heater leads to:

$$\dot{q} = \dot{q}_1 + \dot{q}_2 \tag{9}$$

Equations (7) through (9) are 3 equations in 3 unknowns ($\dot{q}$, $\dot{q}_1$, and $\dot{q}_2$) and can be solved using EES:

```
q_dot_1=(T_h-T_infinity)/(R_cond_ins+(1/R_rad+1/R_conv)^(-1))  "heat transfer to ambient"
q_dot_2=(T_h-T_ice)/R_cond_p                                  "heat transfer to ice"
q_dot=q_dot_1+q_dot_2                                         "energy balance"
```

The temperature at the top of the plate can be computed based on the solution. Update the guess values for the problem (select Update Guess Values from the Calculate menu) and comment out the guessed value for T_{top}:

```
{T_top=360 [K]}                                    "guess for top surface temperature"
```

and calculate T_{top} according to the resistance network:

$$T_{top} = T_h - \dot{q}_1 R_{cond,ins} \tag{10}$$

```
T_top=T_h-q_dot_1*R_cond_ins                       "recalculate top temperature"
```

The result is $\dot{q}_2$ = 1676 W.

The values of the resistances are shown in Figure P1.2-15-2; notice that radiation does not play an important role in the problem because it is a large resistance in parallel with a much smaller one. The resistance to conduction through the plate is also unimportant since it is so small. The resistance to conduction through the insulation and convection are dominant.

b.) How long will it take to melt a hole in the ice?

An energy balance on the ice-to-plate interface leads to:

$$\dot{q}_2 = A_c \, \Delta i_{fus} \, \rho_{ice} \frac{dth_{ice}}{dt} \tag{11}$$

where $\frac{dth_{ice}}{dt}$ is the rate at which the thickness of the ice is reduced. Because there is no energy lost to the water, the rate of ice melting is constant with ice thickness. Therefore the time required to melt the ice is estimated according to:

$$time \frac{dth_{ice}}{dt} = th_{ice} \tag{12}$$

```
q_dot_2=A_c*DELTAi_fus*dth_icedt*rho_ice           "energy balance on ice interface"
dth_icedt*time=th_ice                              "time to melt ice"
time_min=time*convert(s,min)                       "in min"
```

which leads to *time* = 1178 s (19.6 min).

c.) What is the efficiency of the melting process?

The efficiency is defined as the ratio of the energy provided to the plate-to-ice interface to the energy provided to the heater:

$$\eta = \frac{\dot{q}_2}{\dot{q}} \tag{13}$$

```
eta=q_dot_2/q_dot                                         "efficiency of process"
```

which leads to $\eta = 0.93$.

d.) If your battery is rated at 100 amp-hr at 12 V then what fraction of the battery's charge is depleted by running the super ice-auger?

The total amount of energy required to melt a hole in the ice is:

$$Q = \dot{q}\, time \tag{14}$$

The energy stored in the battery ($E_{battery}$) is the product of the voltage and the amp-hr rating. The fraction of the battery charge required is:

$$f = \frac{Q}{E_{battery}} \tag{15}$$

```
Q=q_dot*time                                    "total energy required"
E_battery=100 [amp-hr]*V*convert(A-V-hr,J)      "car battery energy"
f=Q/E_battery                                     "fraction of car battery energy used"
```

which leads to $f = 0.491$.