Problem 1.2-9: Computer Chip Cooling

Computer chips tend to work better if they are kept cold. You are examining the feasibility of maintaining the processor of a personal computer at the sub-ambient temperature of T_{chip} = 0°F. Assume that the operation of the computer chip itself generates $\dot{q}_{chip}$ =10 W of power. Model the processor unit as a box that is a = 2 inch x b = 6 inch x c = 4 inch. Assume that all six sides of the box is exposed to air at T_{air} = 70°F with a convection heat transfer coefficient of $\overline{h}$ =10 W/m^2-K. The box experiences a radiation heat transfer with surroundings that are at T_{sur} = 70°F. The emissivity of the processor surface is ε =0.7 and all six sides experience the radiation heat transfer. You are asked to size the refrigeration system required to maintain the temperature of the processor.

a.) What is the refrigeration load that your refrigeration system must be able to remove to maintain the processor at a steady-state temperature (W)?

The input parameters are entered in EES; notice that the units of each parameter are immediately converted into SI and the units of the associated variables are set (by you) in the Variable Information Window (Figure 2).

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"INPUTS"
T_chip = converttemp(F,K,0)                         "chip temperature"
q_dot_chip = 10 [W]                                 "chip generation"
a = 2 [inch]*convert(inch,m)                        "dimensions of processor"
b = 6 [inch]*convert(inch,m)
c = 4 [inch]*convert(inch,m)
h = 10 [W/m^2-K]                                    "heat transfer coefficient"
T_air=converttemp(F,K,70)                           "air temperature"
T_sur=converttemp(F,K,70)                           "temperature of surroundings"
e = 0.7                                             "emissivity of surface"
```



Variable	Guess	Lower	Upper	Display			Units	Key	Comment
a	0.0508	-infinity	infinity	A	3	N	m		
b	0.1524	-infinity	infinity	A	3	N	m		
c	0.1016	-infinity	infinity	A	1	N	m		
e	0.7	-infinity	infinity	A	3	N	-		
h	10	-infinity	infinity	A	3	N	W/m^2-K		
q_dot_chip	10	-infinity	infinity	A	3	N	W		
T_air	294.3	-infinity	infinity	A	1	N	K		
T_chip	255.4	-infinity	infinity	A	1	N	K		
T_sur	294.3	-infinity	infinity	A	1	N	K		

Figure 2:Variable Information window showing the units for each variable set.

A control volume encompasses just the processor and includes the internal generation from operating the chip ($\dot{q}_{chip}$) as well as convection ($\dot{q}_{conv}$) and radiation ($\dot{q}_{rad}$) and the heat transfer removed by the refrigeration system ($\dot{q}_{load}$). The energy balance is:

$$\dot{q}_{chip} + \dot{q}_{conv} + \dot{q}_{rad} = \dot{q}_{load} \tag{1}$$

The convection and radiation heat transfer rates may be evaluated using the associated rate equations:

$$\dot{q}_{conv} = h\, A_s \left(T_{air} - T_{chip} \right) \tag{2}$$

$$\dot{q}_{rad} = \sigma\, \varepsilon\, A_s \left(T_{sur}^4 - T_{chip}^4 \right) \tag{3}$$

where σ is Stefan-Boltzmann’s constant and A_s is the surface area of the processor:

$$A_s = 2\left(a\,b + b\,c + a\,c \right) \tag{4}$$

These equations are programmed in EES:

```
"part (a)"
A_s=2*(a*b+b*c+a*c)                              "surface area of processor"
q_dot_conv=h*A_s*(T_air-T_chip)                  "convective heat transfer"
q_dot_rad=sigma#*e*A_s*(T_sur^4-T_chip^4)        "radiation heat transfer"
q_dot_chip+q_dot_conv+q_dot_rad=q_dot_load       "energy balance"
```

The units of the variables that have been added are also entered in the Variable Information window (Figure 3).



Variable	Guess	Lower	Upper	Display			Units	Key	Comment
a	0.0508	-infinity	infinity	A	3	N	m		
A_s	0.05677	-infinity	infinity	A	3	N	m^2		
b	0.1524	-infinity	infinity	A	3	N	m		
c	0.1016	-infinity	infinity	A	1	N	m		
e	0.7	-infinity	infinity	A	3	N	-		
h	10	-infinity	infinity	A	3	N	W/m^2-K		
q_dot_chip	10	-infinity	infinity	A	3	N	W		
q_dot_conv	22.08	-infinity	infinity	A	3	N	W		
q_dot_load	39.39	-infinity	infinity	A	3	N	W		
q_dot_rad	7.311	-infinity	infinity	A	3	N	W		
T_air	294.3	-infinity	infinity	A	1	N	K		
T_chip	255.4	-infinity	infinity	A	1	N	K		
T_sur	294.3	-infinity	infinity	A	1	N	K		

Figure 3: Variable Information window with additional units entered.

You can check that your solution is dimensionally consistent by selecting Check Units from the Calculate menu (Figure 4).



Figure 4: Check Units message

Solving the problem (Solve from the Calculate menu) will bring up the Solution Window (Figure 5) and shows that the refrigeration load is 39.4 W.



Figure 5: Solution Window

b.) If the coefficient of performance (COP) of the refrigeration system is nominally 3.5, then how much heat must be rejected to the ambient air (W)? Recall that COP is the ratio of the amount of refrigeration provided to the amount of input power required.

The definition of COP is:

$$\text{COP} = \frac{\dot{q}_{load}}{\dot{w}_{ref}} \tag{5}$$

which is programmed in EES:

```
"part (b)"
COP = 3.5                                        "specified COP"
COP = q_dot_load/w_dot_ref                       "refrigeration power"
```

and solved to show that the refrigeration power will be 11.3 W.

c.) If electricity costs 12¢/kW-hr, how much does it cost to run the refrigeration system for a year, assuming that the computer is never shut off.

The cost of electricity and time of operation are both converted to SI units and used to evaluate the cost per year.

```
"part (c)"
ecost = 12 [cents/kW-hr]*convert(cents/kW-hr,$/J)    "cost of electricity"
time=1 [year]*convert(year,s)                        "time of operation"
cost=time*ecost*w_dot_ref                            "cost of operating system for 1 year"
```

The cost of operating the system for 1 year is $11.8.