Problem 1.6-3: Finned Tube Water Heater

A water heater consists of a copper tube that carries water through hot gas in a furnace, as shown in Fig. P1.6-3(a). The copper tube has an outer radius, $r_{o,tube}$ = 0.25 inch and a tube wall thickness of th = 0.033 inch. The conductivity of the copper is k_{tube} = 300 W/m-K. Water flows through the pipe at a temperature of T_w = 30ºC. The heat transfer coefficient between the water and the internal surface of the pipe is $\overline{h}_w$ = 500 W/m^2-K. The external surface of the tube is exposed to hot gas at T_g = 500ºC. The heat transfer coefficient between the gas and the outer surface of the pipe is $\overline{h}_g$ = 25 W/m^2-K. Neglect radiation from the tube surface.



Figure P1.6-3(a): Copper tube in a water heater.

a.) At what rate is heat is added to the water for a unit length of tube, L = 1 m, for this configuration (W/m)?

The known information is entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
r_o_tube=0.25 [inch]*convert(inch,m)        "outer tube radius"
th=0.033 [inch]*convert(inch,m)             "tube thickness"
k_tube=300 [W/m-K]                          "tube material conductivity"
T_w=converttemp(C,K,30)                     "water temperature"
h_w=500 [W/m^2-K]                           "water to tube heat transfer coefficient"
T_g=converttemp(C,K,500)                    "gas temperature"
h_g=25 [W/m^2-K]                            "gas to tube heat transfer coefficient"
L=1 [m]                                     "unit length of tube"
```

The resistance network that represents this problem is shown in Figure 2.



Figure 2: Resistance network for unfinned tube.

The resistance network includes convection to the inner surface of the tube ($R_{conv,w}$),

$$R_{conv,w} = \frac{1}{h_w\, 2\,\pi \left(r_{o,tube} - th\right) L}, \tag{1}$$

conduction through the tube (R_{cond}),

$$R_{cond} = \frac{\ln\left(\dfrac{r_{o,tube}}{r_{o,tube} - th}\right)}{2\,\pi\, k_{tube}\, L}, \tag{2}$$

and convection from the external surface of the tube ($R_{conv,g}$),

$$R_{conv,g} = \frac{1}{h_g\, 2\,\pi\, r_{o,tube}\, L} \tag{3}$$

The heat transfer is provided by:

$$\dot{q} = \frac{T_g - T_w}{R_{conv,w} + R_{cond} + \left(\dfrac{1}{R_{conv,g}} + \dfrac{1}{R_{rad}}\right)^{-1}} \tag{4}$$

```
"Part a: unfinned tube"
R_conv_w=1/(h_w*2*pi*(r_o_tube-th)*L)          "internal convection resistance"
R_cond=ln(r_o_tube/(r_o_tube-th))/(2*pi*k_tube*L)          "conduction resistance"
R_conv_g=1/(h_g*2*pi*r_o_tube*L)          "external convection"
q_dot=(T_g-T_w)/(R_conv_w+R_cond+R_conv_g)          "heat transfer rate"
```

The heat transfer is $\dot{q}$ =443 W.

b.) What is the dominant resistance to heat transfer in your water heater?

The solution window is shown in Figure 3:



Figure 3:Solution window.

Notice that the values of $R_{conv,g}$ is much higher than R_{cond} or $R_{conv,w}$ and therefore the convection from the surface of the tube limits the heat transfer rate.

In order to increase the capacity of the water heater, you decide to slide washers over the tube, as shown in Fig. P1.6-3(b). The washers are w = 0.06 inch thick with an outer radius of $r_{o,washer}$ = 0.625 inch and have a thermal conductivity of k_{washer} = 45 W/m-K. The contact resistance between the washer and the tube is R''_c = 5x10^{-4} m^2-K/W. The distance between two adjacent washers is b = 0.25 inch.



Figure P1.6-3(b): Water heater with washers installed.

c.) Can the brass washers be treated as extended surfaces (i.e. can the temperature in the washers be considered to be only a function of radius)? Justify your answer with a calculation.

The additional information is entered in EES:

```
"Inputs for finned tube"
w=0.06 [inch]*convert(inch,m)                    "thickness of washers"
r_o_washer=0.625 [inch]*convert(inch,m)          "outer radius of washer"
k_washer=45 [W/m-K]                              "conductivity of washer"
R``_c=5e-4 [m^2-K/W]                             "contact resistance"
b=0.25 [inch]*convert(inch,m)                    "distance between washers"
```

The Biot number associated with the washer must be the ratio of the resistance to conduction from the center of the washer to its edge (axially) to the resistance to heat transfer from its surface:

$$Bi = \frac{R_{cond,x}}{R_{conv,g,x}} \tag{5}$$

where the resistances in Eq. (5) are related to heat transfer axially and so:

$$R_{cond,x} = \frac{w}{2\, k_{washer}\, A_{washer}} \tag{6}$$

$$R_{conv,g,x} = \frac{1}{h_g\, A_{washer}} \tag{7}$$

and

$$A_{washer} = \pi\left(r_{o,washer}^2 - r_{o,tube}^2\right) \tag{8}$$

```
"Part c"
A_washer=pi*(r_o_washer^2-r_o_tube^2)        "exposed area of one side of washer"
R_cond_x=w/(2*k_washer*A_washer)             "conduction resistance axially"
R_conv_g_x=1/(h_g*A_washer)                  "convection resistance axially"
Bi=R_cond_x/R_conv_g_x                       "Biot number"
```

The Biot number is 0.0004 which is much less than one and therefore the temperature gradient in the washer across its thickness is negligible relative to the temperature drop between its surface and the gas. Therefore, the extended surface model is valid and the washer can be treated as a fin.

Assume that your answer to (c) showed that the washers can be treated as extended surfaces and therefore modeled as a fin with the appropriate fin resistance.

d.) Draw a thermal resistance network that can be used to represent this situation. Be sure to draw and label resistances associated with convection through the water ($R_{conv,w}$), conduction through the copper tube (R_{cond}), heat transfer through contact resistance ($R_{contact}$), heat transfer through the washers ($R_{washers}$), and convection to gas from unfinned outer surface ($R_{conv,g,unfinned}$).

The thermal resistance network is shown in Figure 5.



Figure 5: Resistance network representing finned tube.

e.) How much heat is added to the water with the washers installed on the tube for a 1 m length of tube?

The value of $R_{conv,w}$ and R_{cond} do not change from (a). The resistance associated with convection from the unfinned portion of the tube is:

$$R_{conv,g,unfinned} = \frac{1}{h_g\, 2\,\pi\, r_{o,tube}\, L\left(\dfrac{b}{b+w}\right)} \tag{9}$$

where the last term in the denominator is the fraction of the tube surface that is not occupied by the fins. The contact resistance is:

$$R_{contact} = \frac{R''_c}{2\,\pi\, r_{o,tube}\, L\left(\frac{w}{b+w}\right)} \tag{10}$$

where the last term in the denominator is the fraction of the tube surface that is occupied by the fins.

```
"Part d"
R_conv_g_unfinned=1/(h_g*2*pi*r_o_tube*L*(b/(b+w)))  "convection from unfinned surface"
R_contact=R``_c/(2*pi*r_o_tube*L*(w/(b+w)))           "contact resistance"
```

The resistance of the fins is:

$$R_{washers} = \frac{1}{\eta_f\, h_g\, 2\, A_{washer}\left(\frac{L}{b+w}\right)} \tag{11}$$

where the last term in the denominator is the number of washers present on the tube. The fin efficiency, η_f, can be calculated using EES' built-in functions for fin efficiency. Select Function Info from the Options menu and then select the button next to the list at the lower right of the top box. Select Fin Efficiency and then Dimensional Efficiency and scroll over until you find eta_fin_annular_rect (Figure 6).



Figure 6: Fin efficiency function information.

Select Paste and the function call will be inserted into your EES program. Note that Info provides more detailed help about the function. Modify the arguments of the function so that they match your variable names:

```
eta_f=eta_fin_annular_rect(w, r_o_tube, r_o_washer, h_g, k_washer)      "fin efficiency"
```

Use the fin efficiency to calculate the total washer resistance:

```
R_washers=1/(eta_f*h_g*2*A_washer*(L/(b+w)))"washer resistance"
```

The total heat transfer to the finned tube is:

$$\dot{q}_{finned} = \frac{\left(T_g - T_w\right)}{R_{conv,w} + R_{cond} + \left(\frac{1}{R_{conv,g,unfinned}} + \frac{1}{R_{contact} + R_{washers}}\right)^{-1}} \tag{12}$$

```
q_dot_finned=(T_g-T_w)/(R_conv_w+R_cond+(1/R_conv_g_unfinned+1/(R_contact+R_washers))^(-1))
          "total heat transfer rate from finned tube"
```

The addition of the fins has increased the heat transfer rate to 1540 W.