Problem 1.3-9 (1-9 in text): Mass Flow Meter

Figure P1.3-9 illustrates a simple mass flow meter for use in an industrial refinery.



Figure P1.3-9: A simple mass flow meter.

A flow of liquid passes through a test section consisting of an L = 3 inch section of pipe with inner and outer radii, r_{in} = 0.75 inch and r_{out} = 1.0 inch, respectively. The test section is uniformly heated by electrical dissipation at a rate $\dot{g}''' = 1x10^7$ W/m^3 and has conductivity k = 10 W/m-K. The pipe is surrounded with insulation that is th_{ins} = 0.25 inch thick and has conductivity k_{ins} = 1.5 W/m-K. The external surface of the insulation experiences convection with air at T_∞ = 20°C. The heat transfer coefficient on the external surface is $\bar{h}_{out}$ = 20 W/m^2-K. A thermocouple is embedded at the center of the pipe wall. By measuring the temperature of the thermocouple, it is possible to infer the mass flow rate of fluid because the heat transfer coefficient on the inner surface of the pipe ($\bar{h}_{in}$) is strongly related to mass flow rate ($\dot{m}$). Testing has shown that the heat transfer coefficient and mass flow rate are related according to:

$$\bar{h}_{in} = C\left(\frac{\dot{m}}{1\left[\text{kg/s}\right]}\right)^{0.8}$$

where C= 2500 W/m^2-K. Under nominal conditions, the mass flow rate through the meter is $\dot{m}$ = 0.75 kg/s and the fluid temperature is T_f = 18°C. Assume that the ends of the test section are insulated so that the problem is 1-D. Neglect radiation and assume that the problem is steady-state.

a.) Develop an analytical model in EES that can predict the temperature distribution in the test section. Plot the temperature as a function of radial position for the nominal conditions.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
r_out=1.0 [inch]*convert(inch,m)          "outer radius of measurement section"
r_in=0.75 [inch]*convert(inch,m)          "inner radius of measurement section"
h_bar_out=20 [W/m^2-K]                    "external convection coefficient"
```

```
T_infinity=converttemp(C,K,20 [C])          "ambient temperature"
T_f=converttemp(C,K, 18 [C])                "fluid temperature"
k=10 [W/m-K]                                "conductivity"
g```=1e7 [W/m^3]                            "volumetric rate of thermal energy generation"
m_dot=0.75 [kg/s]                           "mass flow rate"
th_ins=0.25 [inch]*convert(inch,m)          "thickness of insulation"
k_ins=1.5 [W/m-K]                           "insulation conductivity"
L= 3 [inch]*convert(inch,m)                 "length of test section"
```

The heat transfer coefficient on the internal surface is computed according to the specified mass flow rate:

```
C=2500 [W/m^2-K]                            "constant for convection relationship"
h_bar_in=C*(m_dot/1 [kg/s])^0.8             "internal convection coefficient"
```

The general solution to a 1-D problem in cylindrical coordinates with constant volumetric thermal energy generation was provided in Table 1-3, to within the unknown constants C_1 and C_2:

$$T = -\frac{\dot{g}''' r^2}{4k} + C_1 \ln(r) + C_2 \tag{1}$$

$$\frac{dT}{dr} = -\frac{\dot{g}''' r}{2k} + \frac{C_1}{r} \tag{2}$$

The boundary condition at the outer edge of the test section is:

$$-k\,2\pi r_{out} L \left(\frac{dT}{dr}\right)_{r=r_{out}} = \frac{\left(T_{r=r_{out}} - T_\infty\right)}{\left(R_{ins} + R_{conv,out}\right)} \tag{3}$$

where R_{ins} is the thermal resistance to conduction through the insulation (provided in Table 1-2):

$$R_{ins} = \frac{\ln\left[\frac{\left(r_{out} + th_{ins}\right)}{r_{out}}\right]}{2\pi L k_{ins}} \tag{4}$$

and $R_{conv,out}$ is the resistance to convection from the outer surface of the insulation:

$$R_{conv,out} = \frac{1}{2\pi\left(r_{out} + th_{ins}\right) L \bar{h}_{out}} \tag{5}$$

```
R_ins=ln((r_out+th_ins)/r_out)/(2*pi*L*k_ins)         "resistance to conduction through insulation"
R_conv_out=1/(2*pi*(r_out+th_ins)*L*h_bar_out)        "resistance to convection from outer surface"
T_r_out=-g```*r_out^2/(4*k)+C_1*ln(r_out)+C_2         "temperature at outer surface of section"
dTdr_r_out=-g```*r_out/(2*k)+C_1/r_out                "temperature gradient at outer surface of section"
```

```
-k*2*pi*r_out*L*dTdr_r_out=(T_r_out-T_infinity)/(R_ins+R_conv_out)
                                                        "boundary condition at r=r_out"
```

The boundary condition at the inner edge of the test section is:

$$\bar{h}_{in}\,2\,\pi\,r_{in}\,L\left(T_f - T_{r=r_{in}}\right) = -k\,2\,\pi\,r_{in}\,L\left(\frac{dT}{dr}\right)_{r=r_{in}} \tag{6}$$

```
T_r_in=-g```*r_in^2/(4*k)+C_1*ln(r_in)+C_2        "temperature at inner surface of section"
dTdr_r_in=-g```*r_in/(2*k)+C_1/r_in               "temperature gradient at inner surface of section"
h_bar_in*2*pi*r_in*L*(T_f-T_r_in)=-k*2*pi*r_in*L*dTdr_r_in
                                                   "boundary condition at r=r_in"
```

The EES code will provide the solution to the constants C_1 and C_2; note that it is not possible to eliminate the unit warnings that are associated with the argument of the natural logarithm in Eq. (1). In fact, if sufficient algebra was carried out, the equations could be placed in a form where the natural logarithm had a dimensionless argument.

The location at which to evaluate the temperature (r) is specified in terms of a dimensionless radial position ($\tilde{r}$) that goes from 0 at the inner surface of the test section to 1 at the outer surface. The temperature is evaluated using Eq. (1):

```
r_bar=0.5 [-]                                  "dimensionless radial position"
r=r_in+r_bar*(r_out-r_in)                      "radial position"
T=-g```*r^2/(4*k)+C_1*ln(r)+C_2                "temperature"
T_C=converttemp(K,C,T)                         "in C"
```

Figure P1.3-9-2 illustrates the temperature as a function of radial position.



Figure 1.3-9-2: Temperature as a function of radius.

b.) Using your model, develop a calibration curve for the meter; that is, prepare a plot of the mass flow rate as a function of the measured temperature at the mid-point of the pipe. The range of the instrument is 0.2 kg/s to 2.0 kg/s.

The dimensionless radial position is set to $\tilde{r}$ =0.5, corresponding to the temperature of the center of the test section. Figure 1.3-9-3 illustrates the mass flow rate through the meter as a function of the measured temperature.



Figure 1.3-9-3: Mass flow rate as a function of the temperature at the center of the pipe wall for several values of the fluid temperature.

The meter must be robust to changes in the fluid temperature. That is, the calibration curve developed in (b) must continue to be valid even as the fluid temperature changes by as much as 10°C.

c.) Overlay on your plot from (b) the mass flow rate as a function of the measured temperature for T_f = 8°C and T_f = 28°C. Is your meter robust to changes in T_f?

The calibration curves generated at T_f = 8°C and T_f = 28°C are also shown in Figure 1.3-9-3. Notice that the fluid temperature has a large effect on the device. For example, if the measured temperature is 80°C then the mass flow rate could be anywhere from 0.45 kg/s to 0.75 kg/s depending on the fluid temperature. The meter is not robust to changes in T_f.

In order to improve the meters ability to operate over a range of fluid temperature, a temperature sensor is installed in the fluid in order to measure T_f during operation.

d.) Using your model, develop a calibration curve for the meter in terms of the mass flow rate as a function of ΔT, the difference between the measured temperatures at the mid-point of the pipe wall and the fluid.

The temperature difference is calculated according to:

$$\Delta T = T_{\tilde{r}=0.5} - T_f \tag{7}$$

```
DT=T-T_f                          "measured temperature difference"
```

Figure 1.3-9-4 illustrates the mass flow rate as a function of the temperature difference:



Figure 1.3-9-4: Mass flow rate as a function of the temperature difference between the measured temperature at the center of the pipe wall and the fluid temperature for several values of the fluid temperature.

e.) Overlay on your plot from (d) the mass flow rate as a function of the difference between the measured temperatures at the mid-point of the pipe wall and the fluid if the fluid temperature is T_f = 8°C and T_f = 28°C. Is the meter robust to changes in T_f?

The calibration curves for T_f = 8°C and T_f = 28°C are also shown in Figure 1.3-9-4; notice that the fluid temperature has almost no effect on the calibration curves and so the meter is robust to changes in the fluid temperature.

f.) If you can measure the temperature difference to within $\delta\Delta T$ = 1 K then what is the uncertainty in the mass flow rate measurement? (Use your plot from part (d) to answer this question.)

The uncertainty in the measured mass flow rate that corresponds to an uncertainty in the temperature difference is evaluated according to:

$$\delta \dot{m} = \left(\frac{\partial \dot{m}}{\partial \Delta T} \right) \delta \Delta T \tag{8}$$

From Figure 1.3-9-4 we see that the partial derivative of mass flow rate with respect to temperature difference decreases with flow rate. At high flow rates (around 2 kg/s), the partial derivative is approximately 0.08 kg/s-K which leads to an uncertainty of 0.08 kg/s. At low flow rates (around 0.2 kg/s), the partial derivative is approximately 0.04 kg/s-K which leads to an uncertainty of 0.04 kg/s.

You can use the built-in uncertainty propagation feature in EES to assess uncertainty automatically.

g.) Set the temperature difference to the value you calculated at the nominal conditions and allow EES to calculate the associated mass flow rate. Now, select Uncertainty Propagation from the Calculate menu and specify that the mass flow rate is the calculated variable while the temperature difference is the measured variable. Set the uncertainty in the temperature difference to 1 K and verify that EES obtains an answer that is approximately consistent with part (f).

The temperature difference is set to 50 K corresponding to approximately the middle of the range of the device. The mass flow rate is commented out and EES is used to calculate the mass flow rate from the temperature difference:

```
DT=50 [K]
{m_dot=0.75 [kg/s]}          "mass flow rate"
```

Select Uncertainty Propagation from the Calculate menu (Figure P1.3-9-5) and select the variable m_dot as the calculated variable and the variable DT as the measured variable.



Figure P1.3-9-5: Determine Propagation of Uncertainty dialog.

Select Set uncertainties and indicate that the uncertainty of the measured temperature difference is 1 K (Figure P1.3-9-6).



Figure P1.3-9-6: Uncertainties of Measured Variables dialog.

Select OK and then then OK again to carry out the calculation. The results are displayed in the Uncertainty Results tab of the Solution window (Figure P1.3-9-7).



Figure P1.3-9-7: Uncertainties Results tab of the Solution window.

The uncertainty calculated by EES is $\delta\dot{m}$ = 0.031 kg/s, which falls between the bounds identified in part (e).

h.) The nice thing about using EES to determine the uncertainty is that it becomes easy to assess the impact of multiple sources of uncertainty. In addition to the uncertainty $\delta\Delta T$, the constant C has an uncertainty of δC = 5% and the conductivity of the material is only known to within δk = 3%. Use EES' built-in uncertainty propagation to assess the resulting uncertainty in the mass flow rate measurement. Which source of uncertainty is the most important?

Select Uncertainty Propagation from the Calculate menu and select the variable m_dot as the calculated variable and the variables DT, C, and k as the measured variables. Set the uncertainty of each of the measured variables according to the problem statement (Figure P1.3-9-8).



Figure P1.3-9-8: Uncertainties of Measured Variables dialog.

The results of the uncertainty calculation are shown in Figure P1.3-9-9.



Figure P1.3-9-9: Uncertainties Results tab of the Solution window.

Notice that the uncertainty has increased to $\delta\dot{m}$ = 0.062 kg/s and that the dominant source of the uncertainty is related to C. The effect of the uncertainty in the conductivity is small (only 5.8% of the total).

i.) The meter must be used in areas where the ambient temperature and heat transfer coefficient may vary substantially. Prepare a plot showing the mass flow rate predicted by your model for ΔT = 50 K as a function of T_∞ for various values of $\bar{h}_{out}$. If the operating range of your meter must include -5°C < T_∞ < 35°C then use your plot to determine the range of $\bar{h}_{out}$ that can be tolerated without substantial loss of accuracy.

Figure P1.3-9-10 illustrates the mass flow rate as a function of T_∞ for various values of $\bar{h}_{out}$.



Figure P1.3-9-10: Mass flow rate predicted with ΔT = 50 K as a function of ambient temperature for various values of the air heat transfer coefficient.

The shaded region in Figure P1.3-9-10 indicates the operating temperature range (in the *x*-direction) and the region of acceptable accuracy (based approximately on the results of part (e)). Figure P1.3-9-10 shows that 5 W/m^2-K < $\bar{h}_{out}$ < 50 W/m^2-K will keep you within the shaded region and therefore this is, approximately, the range of $\bar{h}_{out}$ that can be tolerated without substantial loss of accuracy.