Problem 1.6-6

Your company has developed a technique for forming very small fins on a plastic substrate. The diameter of the fins at their base is D = 1 mm. The ratio of the length of the fin to the base diameter is the aspect ratio, AR = 10. The fins are arranged in a hexagonal close packed pattern. The ratio of the distance between fin centers and to the base diameter is the pitch ratio, PR = 2. The conductivity of the plastic material is k = 2.8 W/m-K. The heat transfer coefficient between the surface of the plastic and the surrounding gas is $\overline{h}$ = 35 W/m^2-K. The base temperature is T_b = 60°C and the gas temperature is T_∞ = 35°C. Do the analysis on a per unit of base area basis (A = 1 m^2).

a.) Determine the number of fins per unit area and the thermal resistance of the unfinned region of the base.

The inputs are entered in EES. An aspect ratio is chosen to start the problem.

```
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
D=1 [mm]*convert(mm,m)                    "diameter"
AR=10 [-]                                 "aspect ratio"
L=AR*D                                    "length"
PR=2 [-]                                  "pitch ratio"
p=PR*D                                    "pitch (distance between fin centers)"
h_bar=35 [W/m^2-K]                        "heat transfer coefficient"
k=2.8 [W/m-K]                             "thermal conductivity"
T_b=converttemp(C,K,60 [C])               "temperature of base"
T_infinity=converttemp(C,K,35 [C])        "temperature of ambient air"
A=1 [m^2]                                      "per unit area of base"
```

The Biot number is:

$$Bi = \frac{\overline{h}\, D}{k} \tag{1}$$

```
Bi=h_bar*D/k                              "Biot number"
```

which leads to Bi = 0.0125, justifying the extended surface approximation. A unit cell of the hexagonal close packed pattern is examined, as shown in Figure 1.



Figure 1: Hexagonal close pack array of fins.

The area of the triangular unit cell is:

$$A_{cell} = \left(\frac{p}{2}\right) p \sin\left(\frac{\pi}{3}\right)\frac{1}{2} \tag{2}$$

The number of fins per unit area is:

$$N'' = \frac{0.25}{A_{cell}} \tag{3}$$

```
A_cell=(p/2)*p*sin(pi/3)/2                "area of unit cell"
N``=0.25/A_cell                           "number of fins per unit area"
```

which leads to N'' = 2.89x10^5 fins/m^2. The number of fins on a 1 m^2 base is:

$$N_{fin} = N'' A \tag{4}$$

The unfinned area of the base is:

$$A_{unfin} = A - N_{fin}\,\pi \frac{D^2}{4} \tag{5}$$

and the thermal resistance of the unfinned region is:

$$R_{unfin} = \frac{1}{\bar{h}\, A_{unfin}} \tag{6}$$

```
N_fin=N``*A                       "number of fins"
A_unfin=A-N_fin*pi*D^2/4          "unfinned surface area"
R_unfin=1/(h_bar*A_unfin)         "thermal resistance of unfinned surface area"
```

which leads to R_{unfin} = 0.0080 K/W.

b.) You have been asked to evaluate whether triangular, parabolic concave, or parabolic convex pin fins will provide the best performance. Plot the heat transfer per unit area of base surface for each of these fin shapes as a function of aspect ratio. Note that the performance of these fins can be obtained from the EES functions eta_fin_spine_parabolic_ND, eta_fin_spine_parabolic2_ND, and eta_fin_spine_triangular_ND. Explain the shape of your plot.

The area of the triangular fins is computed according to Eq. (7) (note that formulae for the fin area can be found in the Help Information for the fin efficiency functions).

$$A_{fin} = N_{fin}\,\pi \frac{D}{2}\sqrt{L^2 + \left(\frac{D}{2}\right)^2} \tag{7}$$

The fin constant is computed according to:

$$mL = \sqrt{\frac{4\,\bar{h}}{k\,D}}\,L \tag{8}$$

The fin efficiency (η) is computed with the eta_fin_spine_triangular_ND.

```
"triangle spine fin"
A_fin=N_fin*pi*D*sqrt(L^2+(D/2)^2)/2          "finned area"
mL=sqrt(4*h_bar/(k*D))*L                      "fin constant"
eta=eta_fin_spine_triangular_ND(mL)           "fin efficiency"
```

The fin resistance is computed according to:

$$R_{fin} = \frac{1}{\bar{h}\,A_{fin}\,\eta} \tag{9}$$

The total resistance to heat transfer is:

$$R_{total} = \left(\frac{1}{R_{fin}} + \frac{1}{R_{unfin}}\right)^{-1} \tag{10}$$

and the rate of heat transfer is:

$$\dot{q} = \frac{\left(T_b - T_\infty\right)}{R_{total}} \tag{11}$$

```
R_fin=1/(h_bar*A_fin*eta)              "thermal resistance of finned surface area"
R_total=(1/R_fin+1/R_unfin)^(-1)       "total thermal resistance"
q_dot=(T_b-T_infinity)/R_total         "total heat transfer"
```

which leads to $\dot{q}$ = 3128 W/m^2. Figure 2 illustrates the rate of heat transfer for triangular pin fins as a function of aspect ratio. Note that increasing the aspect ratio increases the heat transfer because the surface area is larger for longer fins. However, as the aspect (and therefore the length) increases, the fins become less efficient.



Figure 2: Rate of heat transfer per area as a function of the aspect ratio for various fin shapes.

In order to analyze the concave parabolic fins, the formulae for the fin area, fin constant, and efficiency are commented out and replaced with:

```
{"triangle spine fin"
A_fin=N_fin*pi*D*sqrt(L^2+(D/2)^2)/2                          "finned area"
mL=sqrt(4*h_bar/(k*D))*L                                      "fin constant"
eta=eta_fin_spine_triangular_ND(mL)                           "fin efficiency"}
"concave parabolic fin"
C_3=1+2*(D/L)^2
C_4=sqrt(1+(D/L)^2)
A_fin=N_fin*pi*L^3*(C_3*C_4-L*ln(2*D*C_4/L+C_3)/(2*D))/(8*D)  "finned area"
mL=sqrt(4*h_bar/(k*D))*L                                      "fin constant"
eta=eta_fin_spine_parabolic_ND(mL)                            "fin efficiency"
```

The performance of the concave parabolic fins is overlaid onto Figure 2. In order to analyze the convex parabolic fins, these equations are replaced with:

```
{"concave parabolic fin"
C_3=1+2*(D/L)^2
C_4=sqrt(1+(D/L)^2)
A_fin=N_fin*pi*L^3*(C_3*C_4-L*ln(2*D*C_4/L+C_3)/(2*D))/(8*D)  "finned area"
mL=sqrt(4*h_bar/(k*D))*L                                      "fin constant"
eta=eta_fin_spine_parabolic_ND(mL)                            "fin efficiency"}
"convex parabolic"
A_fin=N_fin*pi*D^4*((4*L^2/D^2+1)^1.5-1)/(6*L^2)              "finned area"
mL=sqrt(4*h_bar/(k*D))*L                                      "fin constant"
eta=eta_fin_spine_parabolic2_ND(mL)                           "fin efficiency"
```

The performance of convex parabolic fins is also overlaid onto Figure 2. Figure 2 shows that the convex pin fin provides the best performance; this is because this type of fin has the most surface area.

c.) Assume that part (b) indicated that parabolic convex fins are the best. You have been asked whether it is most useful to spend time working on techniques to improve (increase) the aspect ratio or improve (reduce) the pitch ratio. Answer this question using a contour plot that shows contours of the heat transfer per unit area in the parameter space of *AR* and *PR*.

The aspect ratio and pitch ratio are commented out and a parametric table is created with these variables and the total heat transfer. A contour plot is shown in Figure 3 and shows that it is much more beneficial to reduce the pitch ratio than the aspect ratio.



Figure 3: Contour plot of refrigeration per unit area as a function of pitch ratio and aspect ratio.