Problem 1.7-3 (1-15 in text): Material Processing

Figure P1.7-3 illustrates a material processing system.



Figure P1.7-3: Material processing system.

Material is extruded and enters the oven at T_{in} = 300 K with velocity u = 0.75 m/s. The material has velocity u = 0.75 m/s and diameter D = 5 cm. The conductivity of the material is k = 40 W/m-K and the thermal diffusivity is α = 0.001 m^2/s.

In order to precisely control the temperature of the material, the oven wall is placed very close to the outer diameter of the extruded material and the oven wall temperature distribution is carefully controlled. The gap between the oven wall and the material is th = 0.6 mm and the oven-to-material gap is filled with gas that has conductivity k_g = 0.03 W/m-K. Radiation can be neglected in favor of convection through the gas from the oven wall to the material. For this situation, the heat flux experienced by the material surface can be approximately modeled according to:

$$\dot{q}''_{conv} \approx \frac{k_g}{th}\left(T_w - T\right)$$

where T_w and T are the oven wall and material temperatures at that position. The oven wall temperature varies with position x according to:

$$T_w = T_f - \left(T_f - T_{w,0}\right)\exp\left(-\frac{x}{L_c}\right)$$

where $T_{w,0}$ is the temperature of the wall at the inlet (at x = 0), T_f = 1000 K is the temperature of the wall far from the inlet, and L_c is a characteristic length that dictates how quickly the oven wall temperature approaches T_f. Initially, assume that $T_{w,0}$ = 500 K, T_f = 1000 K, and L_c = 1 m. Assume that the oven can be approximated as being infinitely long.

a.) Is an extended surface model appropriate for this problem?

The inputs are entered in EES:

```
$UnitSystem SI MASS  DEG PA C J
```

```
$Tabstops 0.2 0.4 0.6 0.8 3.5

k=40 [W/m-K]                                "conductivity"
u=0.75 [m/s]                                "velocity"
T_f=1000 [K]                                "wall temperature far from the inlet"
T_w_0=500 [K]                               "wall temperature at the inlet"
L_c=1 [m]                                   "characteristic length which oven wall approaches T_f"
T_in=300 [K]                                "inlet temperature"
alpha=0.001 [m^2/s]                         "thermal diffusivity"
k_g=0.03 [W/m-K]                            "gas conductivity"
th=0.6 [mm]*convert(mm,m)                   "oven-to-material gap thickness"
D=5 [cm]*convert(cm,m)                      "diameter"
```

The Biot number is the ratio of the resistance that is neglected (internal conduction) to the resistance that is considered (conduction across the gap):

$$Bi = \frac{k_g}{th}\frac{D}{2\,k} \tag{1}$$

```
Bi=(k_g/th)*D/(2*k)                   "Biot number"
```

which leads to Bi = 0.031. This is sufficiently less than 1 to justify an extended surface model.

b.) Assume that your answer to (a) was yes. Develop an analytical solution that can be used to predict the temperature of the material as a function of *x*.

An energy balance on a control volume differential for a differential (in *x*) segment of the material is shown in Figure P1.7-3-2.



Figure P1.7-3-2: Energy balance on a differential control volume.

The energy balance suggested by Figure P1.7-3-2 is:

$$\left(\rho\,u\,A_c\,c\,T\right)_x + \left(-k\,A_c\frac{dT}{dx}\right)_x = \left(\rho\,u\,A_c\,c\,T\right)_{x+dx} + \left(-k\,A_c\frac{dT}{dx}\right)_{x+dx} + \frac{k_g}{th}\,per\,dx\left(T-T_w\right) \tag{2}$$

where c is the specific heat capacity, A_c is the cross-sectional area and *per* is the perimeter of the material:

$$A_c = \pi \frac{D^2}{4} \tag{3}$$

$$per = \pi D \tag{4}$$

```
A_c=pi*D^2/4                                    "cross-sectional area"
per=pi*D                                        "perimeter"
```

Expanding the terms in Eq. (2) and simplifying:

$$0 = \rho u A_c c \frac{dT}{dx} - k A_c \frac{d^2T}{dx^2} + \frac{k_g}{th} per\left(T - T_w\right) \tag{5}$$

Rearranging Eq. (5) and dividing through by $k A_c$ leads to:

$$\frac{d^2T}{dx^2} - \frac{u}{\alpha}\frac{dT}{dx} - \frac{k_g \, per}{th \, k \, A_c} T = -\frac{k_g \, per}{th \, k \, A_c} T_w \tag{6}$$

Substituting the wall temperature variation into Eq. (6) leads to:

$$\frac{d^2T}{dx^2} - \frac{u}{\alpha}\frac{dT}{dx} - m^2 T = -m^2 \left[T_f - \left(T_f - T_{w,0}\right)\exp\left(-\frac{x}{L_c}\right)\right] \tag{7}$$

where

$$m = \sqrt{\frac{k_g \, per}{th \, k \, A_c}} \tag{8}$$

```
m=sqrt(4*k_g/(th*k*D))                "fin parameter"
```

The boundary conditions are the inlet temperature:

$$T_{x=0} = T_{in} \tag{9}$$

and the temperature must approach T_f as x approaches infinity:

$$T_{x \to \infty} = T_f \tag{10}$$

The solution is broken into a homogeneous and particular component:

$$T = T_h + T_p \tag{11}$$

and substituted into Eq. (7):

$$\underbrace{\frac{d^2T_h}{dx^2}-\frac{u}{\alpha}\frac{dT_h}{dx}-m^2\,T_h}_{\text{homogeneous ordinary differential equation}}+\underbrace{\frac{d^2T_p}{dx^2}-\frac{u}{\alpha}\frac{dT_p}{dx}-m^2\,T_p=-m^2\left[T_f-\left(T_f-T_{w,0}\right)\exp\left(-\frac{x}{L_c}\right)\right]}_{\text{particular ordinary differential equation}} \tag{12}$$

The solution to the homogeneous differential equation is:

$$T_h=C_1\exp\left[\left(\frac{u+\sqrt{u^2+4\alpha^2\,m^2}}{2\alpha}\right)x\right]+C_2\exp\left[\left(\frac{u-\sqrt{u^2+4\alpha^2\,m^2}}{2\alpha}\right)x\right] \tag{13}$$

The particular solution is obtained by the method of undetermined coefficients; the assumed form of the particular solution is:

$$T_p=C_3\exp\left(-\frac{x}{L_c}\right)+C_4 \tag{14}$$

and substituted into the particular differential equation:

$$\frac{C_3}{L_c^2}\exp\left(-\frac{x}{L_c}\right)+\frac{u}{\alpha\,L_c}C_3\exp\left(-\frac{x}{L_c}\right)-m^2C_3\exp\left(-\frac{x}{L_c}\right)-m^2\,C_4=-m^2\left[T_f-\left(T_f-T_{w,0}\right)\exp\left(-\frac{x}{L_c}\right)\right] \tag{15}$$

Equation (15) provides one equation for C_3 that is obtained by considering the exponential terms:

$$C_3=\frac{m^2\left(T_f-T_{w,0}\right)}{\left(\frac{1}{L_c^2}+\frac{u}{\alpha\,L_c}-m^2\right)} \tag{16}$$

and another equation for C_4 that is obtained by considering the constant terms:

$$C_4=T_f \tag{17}$$

Substituting Eqs. (13), (14), (16), and (17) leads to:

$$\begin{aligned}T=C_1\exp\left[\left(\frac{u+\sqrt{u^2+4\alpha^2\,m^2}}{2\alpha}\right)x\right]+C_2\exp\left[\left(\frac{u-\sqrt{u^2+4\alpha^2\,m^2}}{2\alpha}\right)x\right]\\+\frac{m^2\left(T_f-T_{w,0}\right)}{\left(\frac{1}{L_c^2}+\frac{u}{\alpha\,L_c}-m^2\right)}\exp\left(-\frac{x}{L_c}\right)+T_f\end{aligned} \tag{18}$$

The constants C_1 and C_2 are obtained by considering the boundary conditions. Substituting Eq. (18) into Eq. (10) leads to:

$$C_1 \exp\left[\left(\frac{u+\sqrt{u^2+4\alpha^2 m^2}}{2\alpha}\right)\infty\right]+T_f = T_f \tag{19}$$

which can only be true if $C_1 = 0$. Therefore:

$$T = C_2 \exp\left[\left(\frac{u-\sqrt{u^2+4\alpha^2 m^2}}{2\alpha}\right)x\right]+\frac{m^2\left(T_f - T_{w,0}\right)}{\left(\frac{1}{L_c^2}+\frac{u}{\alpha L_c}-m^2\right)}\exp\left(-\frac{x}{L_c}\right)+T_f \tag{20}$$

Substituting Eq. (20) into Eq. (9) leads to:

$$C_2 + \frac{m^2\left(T_f - T_{w,0}\right)}{\left(\frac{1}{L_c^2}+\frac{u}{\alpha L_c}-m^2\right)}+T_f = T_{in} \tag{21}$$

or

$$C_2 = T_{in} - T_f - \frac{m^2\left(T_f - T_{w,0}\right)}{\left(\frac{1}{L_c^2}+\frac{u}{\alpha L_c}-m^2\right)} \tag{22}$$

```
C_2=T_in-T_f-m^2*(T_f-T_w_0)/(1/L_c^2+u/(alpha*L_c)-m^2)     "boundary condition at x=0"
```

The solution for the material temperature and the wall temperature are entered in EES:

```
x=0.5 [m]                                                     "position"
T=C_2*exp(((u-sqrt(u^2+4*alpha^2*m^2))/(2*alpha))*x)+m^2*(T_f-T_w_0)*&
   exp(-x/L_c)/(1/L_c^2+u/(alpha*L_c)-m^2)+T_f                "temperature of the material"
T_w=T_f-(T_f-T_w_0)*exp(-x/L_c)                               "wall temperature"
```

c.) Plot the temperature of the material and the temperature of the wall as a function of position for $0 < x < 20$ m. Plot the temperature gradient experienced by the material as a function of position for $0 < x < 20$ m.

Figure P1.7-3-3 illustrates the temperature of the material and the wall as a function of position.



Figure P1.7-3-3: Temperature of the material and the wall as a function of position.

The temperature gradient is evaluated by differentiating Eq. (20):

$$\frac{dT}{dx} = C_2 \left(\frac{u - \sqrt{u^2 + 4\alpha^2 m^2}}{2\alpha} \right) \exp\left[\left(\frac{u - \sqrt{u^2 + 4\alpha^2 m^2}}{2\alpha} \right) x \right] - \frac{m^2 \left(T_f - T_{w,0} \right)}{L_c \left(\frac{1}{L_c^2} + \frac{u}{\alpha L_c} - m^2 \right)} \exp\left(-\frac{x}{L_c} \right) \quad (23)$$

```
dTdx=C_2*((u-sqrt(u^2+4*alpha^2*m^2))/(2*alpha))*exp(((u-sqrt(u^2+4*alpha^2*m^2))/(2*alpha))*x)&
    -m^2*(T_f-T_w_0)*exp(-x/L_c)/(1/L_c^2+u/(alpha*L_c)-m^2)/L_c "temperature gradient"
```

Figure P1.7-3-4 illustrates the temperature gradient as a function of position.



Figure P1.7-3-4: Temperature gradient in the material as a function of position.

The parameter L_c can be controlled in order to control the maximum temperature gradient experienced by the material as it moves through the oven.

d.) Prepare a plot showing the maximum temperature gradient as a function of L_c. Overlay on your plot the distance required to heat the material to T_p = 800 K (L_p). If the maximum temperature gradient that is allowed is 60 K/m then what is the appropriate value of L_c and the corresponding value of L_p.

The value L_p is obtained:

```
T_p=800 [K]
T_p=C_2*exp(((u-sqrt(u^2+4*alpha^2*m^2))/(2*alpha))*L_p)+&
        m^2*(T_f-T_w_0)*exp(-L_p/L_c)/(1/L_c^2+u/(alpha*L_c)-m^2)+T_f
```

which leads to L_p = 10.18 m.

The maximum temperature gradient can be obtained by using EES' optimization routines. Setup a parametric table that includes the variables L_c, x, dTdx, L_p, and L_c. The value of L_c that is set in the Equations window is commented out and the values of L_c in the table are varied from 0.1 to 5 m. Min/Max Table is selected from the Calculate menu. The value of dTdx is maximized by varying x with bounds from 0 to some large value. The maximum temperature gradient and value of L_p are shown Figure P1.7-3-5 as a function of L_c. Figure P1.7-3-5 indicates that L_c should be equal to 1.8 m in order to control the temperature gradient, which leads to L_p = 11 m.



Figure P1.7-3-5: Maximum temperature gradient and L_p as a function of L_c.